担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

この講義について

作成日： October 6, 2011　Version：1.1

担当教員：
川平 友規（かわひら ともき，講義担当，A441 号室，`kawahira@math.nagoya-u.ac.jp`）
佐藤 猛（さとう たけし，演習担当，A359 号室，`sato@math.nagoya-u.ac.jp`）

講義のウェブサイト：
`http://www.math.nagoya-u.ac.jp/~kawahira/courses/11W-keisan.html`
配布されたプリントが pdf 形式でダウンロードできます．また，毎週の進捗状況についてコメントしていきます．

本講義の目的（コースデザインより）： 本講義の目的は，数理科学の問題に対してコンピュータを活用するための基礎知識を習得することである．具体的には，数式処理ソフトウェア *Mathematica* を用いて，数理科学の諸問題に取り組む．

講義日と授業内容（予定）：

第 1 回	10/6	**基礎編**（*Mathematica* 入門）：
第 2 回	10/13	端末の設定，代数（記号）演算，方程式，
第 3 回	10/20	数列，リストの操作，微分・積分，
第 4 回	10/27	線形代数，平面グラフィックス，
第 5 回	11/10	空間グラフィックス
第 6 回	11/17	**応用編**（「可視化」をテーマに）：
第 7 回	11/24	テイラー展開，フーリエ展開，
第 8 回	12/1	ベクトル場，等位面，微分方程式，
第 9 回	12/8	複素関数，乱数，ブラウン運動，
第 10 回	12/15	パーコレーション，素数とゼータ
第 11 回	12/22	**発展編**（プログラミングの初歩）：
第 12 回	1/12	計算誤差の問題，ループ（`Do`，`For`，`While`）
第 13 回	1/19	力学系，カオス，フラクタル
第 14 回	1/26	連分数，セル・オートマトン

講義の進め方： 1 限は教室での講義，2 限は下記の端末室での実習となります．

- 講義：理 1 号館 109 号室
- 演習：工学部 7 号館 4 階情報メディア教育センター（主センター）・端末室 B

1 限の最初に出席確認を兼ねたクイズ（小テスト）を行います．また 2 限の実習中に，その日の課題をメールに添付し提出（送信）してください．

成績の評価方法： 1・2 限の出席とクイズ，課題を総合的に判断して評価します．(1 限の遅刻は 3 回目以降欠席とカウント．欠席 5 回で単位放棄とみなします．)

課題の提出方法： 演習で出される課題は，次の要領でメールの添付ファイルとして送信してください．
● **送信先：**`comp0-2011@math.nagoya-u.ac.jp`（`comp0` の `0` はゼロ．川平と佐藤に届きます.）
● **締切：**2 限の講義終了時．その後は一切受け付けません．
● **メールの様式：**メールの件名と添付ファイル名はともに半角英数字で「講義の回数-学籍番号」

の形とします．たとえば，第 5 回目の課題を学籍番号 060123456 の人が提出するときは
件名： `05-060123456`
添付ファイル名： `05-060123456.m` （ファイル形式については後述）
のようにすること．

!! 注意 !! 不正行為を防ぐために，学生間のコンピュータファイルの受け渡しは禁止します．
その他，不正行為には厳しく対処するので注意すること．

その他：

(1) 学内のコンピュータの利用法に関する情報は情報メディア教育システムのページ `http://www.media.nagoya-u.ac.jp/` にまとめられています．このページは「ブックマーク」に登録しておくとよいでしょう．

(2) *Mathematica* を自習・活用したいとき：平日の 8:45—17:00 の間，各端末室のパソコンは講義に使われている時間を除いて自由に利用することができます．たとえば今学期，上記の時間帯であれば主センターの端末室 A・B のいずれかが利用できる予定です．詳しくは `http://www.media.nagoya-u.ac.jp/lectures/` の講義予定表を参照してください．他の端末室の利用スケジュールも確認できます．

参考書：

- 日本 *Mathematica* ユーザー会，「入門 *Mathematica*」（東京電機大学出版局）
- 榊原進，「はやわかり *Mathematica*」（共立出版）

オフィスアワー： 授業中・授業前後の質問は大歓迎です．それ以外の時間に質問したい場合は，ぜひオフィスアワー（教員ごとの質問受付時間）を活用してください．私のオフィスアワーは，Cafe David（カフェ・ダヴィド）という合同のオフィスアワー内に設定しています．Cafe David は月曜から金曜の 12:00-13:30 にオープンし，コーヒー・紅茶を無料で提供しています．私の担当は**月曜日**です．その他のスタッフ（名札をつけています）も待機しているので，自由に質問していただいてかまいません．数理学科の学部生・院生もたくさんいるので，勉強のこと，進路のことなど，情報収集の場としても活用してください．

謝辞： 授業を準備するにあたって，2008-2009 年度にこの講義を担当された糸健太郎さん，2010 年度に担当された粟田英資さんにはたいへん貴重なアドバイスをいただきました．とくにこのシラバスは糸さんが授業で使われたものをベースにしております．また内藤久資さんにはシステム関連でご協力いただきました．みなさん，どうもありがとうございました．

よく使う記号など：数の集合

(1) $\mathbb{C}$: 複素数全体	(2) $\mathbb{R}$: 実数全体	(3) $\mathbb{Q}$: 有理数全体
(4) $\mathbb{Z}$: 整数全体	(5) $\mathbb{N}$: 自然数全体	(6) $x \in \mathbb{R}$: x は実数

ギリシャ文字

(1) α: アルファ	(2) β: ベータ	(3) γ, Γ: ガンマ	(4) δ, Δ: デルタ	(5) ϵ: イプシロン
(6) ζ: ゼータ	(7) η: エータ	(8) θ, Θ: シータ	(9) ι: イオタ	(10) κ: カッパ
(11) λ, Λ: ラムダ	(12) μ: ミュー	(13) ν: ニュー	(14) ξ, Ξ: クシー	(15) o: オミクロン
(16) π, Π: パイ	(17) ρ: ロー	(18) σ, Σ: シグマ	(19) τ: タウ	(20) υ, Υ: ウプシロン
(21) ϕ, Φ: ファイ	(22) χ: カイ	(23) ψ, Ψ: プサイ	(24) ω, Ω: オメガ	

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

Mathematica 入門

作成日： October 6, 2011　Version：1.3

今日やること

(1) **Mac にログインする．**

Mac が英語環境になっているときは，次のようにして日本語環境に変更する：

メニューバーの「アップルマーク」→[System Preference...]→[International] この中の Languages の中の [日本語] を [English] の上に移動させる．このウィンドウを閉じて，ログアウトし，再度ログインする．

(2) **メールソフト Thunderbird の設定をする．**

(a) まず Dock [1]のアイコンをクリックして Thunderbird を起動する．アイコンが Dock に無い場合は,「TEMPHDD」→「アプリケーション」→Thunderbird.app と辿る．このアイコンは Dock の中にドラッグ&ドロップしておくとよい

(b) Thunderbird でメールの送受信設定をする．設定の方法は別紙参照のこと．

(c) 設定が終わったら，まず自分にテストメールを送ってみる．

(d) 自分へのメールが無事に送信・受信できること確認したら，次に
`comp0-2011@math.nagoya-u.ac.jp`
にテストメールを送る．メールの件名は「00-060123456」（半角で．00-のあとは自分の学籍番号にかえる）の形とし，本文に名前（ふりがな）を書く．

(3) ***Mathematica*** **の起動，設定，実行．**

(a) *Mathematica* を起動する．*Mathematica* のアプリケーション本体は Thunderbird と同じところにある.（このアイコンも Dock の中にドラッグ&ドロップしておくとよい.）

(b) *Mathematica* が日本語環境でない場合：メニューバーの「Mathematica」→「Preferences」→「Interface」で,「Language for menus and dialog boxes」にあるドロップメニューから「Japanese」を選ぶ．さらに *Mathematica* を一度終了し，再度起動する．

(c) *Mathematica* の文字（フォント）を適宜大きくする（これは暇なときにやればよい)：メニューバーの「*Mathematica*」→「環境設定. .」→「詳細」→「オプションインスペクタを開く」→「書式設定」→「フォント設定」→「フォントサイズ」これを 20 ∼ 30 ぐらいの数にする．

(d) 次ページの課題をがんばる．

(e) *Mathematica* の「名称未定義-1」というファイルをメニューバーの「ファイル」→「別名で保存...」として，01-060123456.nb のような名前で保存する．保存場所は「TEMPHDD」→（家のアイコン）→「書類」の中とする．必要に応じて新規フォルダを作ってもよい.（*Mathematica* はまだ閉じちゃだめ.）

(f) 次に同じファイルを今度はパッケージ形式で保存する：すなわち,「別名で保存...」から「フォーマット」のタブで *Mathematica* パッケージ (*.m) を選び，01-060123456.m という名前にして書類フォルダに保存する．

[1]画面下のアイコンが並んでいるところ

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

課題を提出する際の注意 !! 通常はノートブック形式（*.nb）で保存するが，このままだとファイルサイズが非常に大きいので，課題としてメールに添付する際にはパッケージ形式（*.m）で保存してこちらを送ること．

(g) 課題の提出:この 01-060123456.m をメールに添付し comp0-2011@math.nagoya-u.ac.jp に送る．ただしメールの件名は「01-060123456」とすること．

今日の課題 (10/6).

右側の式を 1 行ずつ入力・評価（evaluate）しノートブックを作成せよ．
式の評価は Shift ＋ Return （Shift を押しながら Return）である．

四則と定数

和：	`1 + 1`
積と商：	`15*(10 + 3)/12`
分数：	`1/3 + 1/12`
べき乗：	`3^10`
階乗：	`100!`
定数 π：	`Pi`
π の数値化：	`N[Pi]`
π を 100 桁：	`N[Pi, 100]`
1000 番目の素数：	`Prime[1000]`

文字式と方程式

$(a+b)^{10}$ の展開，結果を q とおく ：	`q = Expand[(a + b)^10]`
q をもう一度因数分解 ：	`Factor[q]`
方程式を解く：	`Solve[x^2 - x - 2 == 0, x]`
文字が入っても OK ：	`Solve[x^2 - x + c == 0, x]`
1 の 10 乗根：	`Solve[x^10 == 1, x]`
同じ方程式を数値的に解く：	`NSolve[x^10 == 1, x]`

微分・積分

$\sin x$ を x で微分：	`D[Sin[x], x]`
$\cos xy^2$ を y で偏微分：	`D[Cos[x*y^2], y]`
$\cos xy^2$ を x で偏微分：	`D[Cos[x*y^2], x]`
定積分 $\int_0^1 x^2\,dx$ ：	`Integrate[x^2, {x, 0, 1}]`

グラフ

$\sin x$ の $-\pi \le x \le \pi$ でのグラフ： `Plot[Sin[x], {x, -Pi, Pi}]`
$\sin(e^x)$ の $-5 \le x \le 5$ でのグラフ: `Plot[ Sin[Exp[x]], {x, -5, 5}]`
$\sin\sqrt{x^2+y^2}$ の $-10 \le x, y \le 10$ でのグラフ：
`Plot3D[ Sin[Sqrt[x^2 + y^2]], {x, -10, 10}, {y, -10, 10}]`

研究

- 課題が提出できたら，ノートブックをさらに別名で保存し，数値や関数を変えてどんな結果がでるか試してみよ．
- ヘルプ ⟶ ドキュメントセンターと辿ってドキュメントセンターを開き，*Mathematica* の機能についていろいろ勉強せよ．
- ドキュメントセンターの検索ウインドウから `N` `Plot` `Sin` など，今日使ったコマンドを検索してみよ．

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

式の変形と値

作成日： October 13, 2011　Version : 1.2

役立つキー操作

やり直し：	Command + Z	式の評価（実行）：	Shift + Return
コピー：	Command + C	前の入力をコピー ：	Command + L
貼り付け：	Command + V	単語を補完：	Command + K
切り取り：	Command + X	根号 $\sqrt{\ }$ の入力：	Command + 2

注． Windows や UNIX の場合 Command は Ctrl となる．

入力時の注意

(1) 必ず半角英数字で入力.（全角漢字も使えるが混乱のもとになる.）

(2) 入出力を消したいときはノートブック右側の] をマウス等で選択し，Delete.

(3) 数の掛け算は `x*y` でも `x y`（間に半角空白をはさむ）でもよい．

(4) 計算の優先順位を表す括弧は常に ( )：`1/{1 + (2 + 3)}` ではなく，`1/(1 + (2 + 3))`. `{}` はリストやベクトル（`{a,b,c}`など）に，`[]` は関数（`Sin[x]` など）に用いられる．

(5) *Mathematica* 内ですでに定義されている関数（`Sin[x]`, `Exp[x]` など）は大文字で始まる．

(6) *Mathematica* は一度代入された値をずっと覚えている．もし `a = 1; b = a + 1` と入力・評価されたら，そのあと a と b の値はそれぞれ 1 と 2 であり続ける．再度別の意味で使うときは，かならず値をクリアしよう：`Clear[a, b]` もしくは `a = .; b = .`

(7) 直前の出力は `%` で参照できるが，混乱のもとになるので課題には使わないこと．

基本的な関数と定数（関数）

注． 下記の定数は「厳密値」として記号のまま扱われる．近似値がほしいときは N 関数などで適宜数値化する．

円周率（定数）：	`Pi`	指数関数：	`Exp[ ]`
自然対数の底（定数)）：	`E`	自然対数 $\log a$：	`Log[a]`
虚数単位（定数）：	`I`	対数 $\log_b a$ ：	`Log[b , a]`
黄金比（定数）：	`GoldenRatio`	数値化：	`N[ ]`
平方根：	`Sqrt[ ]`	m 桁で数値化 ：	`N[ , m]`
三角関数：	`Sin[ ]`	x 以下の最大の整数 ：	`Floor[ ]`
	`Cos[ ]`	$a, b, c, \ldots$ の最大値 ：	`Max[a,b,c,...]`
	`Tan[ ]`	$a, b, c, \ldots$ の最小値 ：	`Min[a,b,c,...]`
逆三角関数：	`ArcSin[ ]` など		

注意． 以下とくに断らないかぎり,「近似値」といったら N 関数による値とする．

問題 1. *Mathematica* を用いて計算せよ．

(1) π^e, e^π , $\pi + 20$ の近似値をそれぞれ有効数字 7 桁で求め，もっとも小さなものを求めよ．

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

(2) $a=\sqrt{2+\sqrt{3}}$, $b=\sqrt{15}-2$, . $c=\dfrac{1}{\sqrt{5}+\sqrt{3}}$, $d=\sqrt{3}$ とする．これらの近似値をもとめ，最大のものと最小のものを決定せよ． **（神戸薬大・改）**

(3) $a=\sqrt{3}+\sqrt{2}i$, $b=\sqrt{3}-\sqrt{2}i$ のとき，$c=\dfrac{a}{b}+\dfrac{b}{a}$ と $d=a^4-2a^2$ それぞれの厳密値および有効数字 20 桁の近似値を求めよ． **（'98 センター追試・改）**

(4) $\sqrt{42+12\sqrt{6}}$ の整数部分を p，小数部分を q とするとき，p, q, $r=\dfrac{p}{q(q+4)}$ の値を厳密値および近似値で求めよ． **（成蹊大・改）**

文字式の操作

問題 2. 順に入力して結果を確かめよ．

整数係数での**因数分解**：	`Factor[x^10-1]`
式の展開，結果を s とせよ：	`s = Expand[(1 + x + y)^3]`
s を x で**まとめる**：	`Collect[s, x]`
s に $x=2$ を**代入**：	`s/.{x -> 2}`
s に $x=2, y=-3$ を**代入**：	`s/.{x -> 2, y -> -3}`
$t0=\dfrac{1}{x^3-1}$ とおく：	`t0 = 1/(x^3-1)`
$t0$ を**部分分数に分解**，結果を $t1$ とおく：	`t1 = Apart[t0]`
$t1$ を**通分**，結果を $t2$ とおく：	`t2 = Together[t1]`
$t2$ の式の**分母だけ展開する**：	`ExpandDenominator[t2]`
$t2$ の式を**簡単にする**：	`Simplify[t2]`
`Simplify` で簡単に（単なる計算）：	`Simplify[Sqrt[512]/4]`
`Simplify` で簡単に（公式を適用）：	`Simplify[Sin[x]^2 + Cos[x]^2]`
`FullSimplify` はもっとすごい：	`FullSimplify[Cos[x] + I Sin[x]]`

問題 3. *Mathematica* を用いて計算せよ．

(1) **問題 1** の (3)(4) の入出力部分にもどり，c, d, q, r の厳密値に `Simplify`（もしくは `FullSimplify`）を施せ．

(2) $x^4-3x^2y^2+y^4$ を因数分解せよ．その結果に，$x=-1$ を代入せよ． **（04 名古屋経大・改）**

(3) $(x+y)(y+z)(z+x)+xyz$ を展開し，さらに因数分解せよ．その式に $x=\sqrt{2}$ および $y=\sqrt{3}$ を代入せよ． **（00 名城大・改）**

(4) $\dfrac{1}{(x+1)(x+2)(x+3)}=\dfrac{a}{x+1}+\dfrac{b}{x+2}+\dfrac{c}{x+3}$ が恒等式であるとき，a, b, c の値を求めよ． **（97 静岡理工科大）**

(5) $\dfrac{1}{x+2}+\dfrac{2x}{(x+2)^2}+\dfrac{3x^2}{(x+2)^3}=\dfrac{a}{x+2}+\dfrac{b}{(x+2)^2}+\dfrac{c}{(x+2)^3}$ が恒等式であるとき，a, b, c の値を求めよ． **（92 久留米工大）**

(6) $\dfrac{a^2}{(a-b)(a-c)}+\dfrac{b^2}{(b-c)(b-a)}+\dfrac{c^2}{(c-a)(c-b)}$ を簡単にせよ． **（02 横浜市大・改）**

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 02-060123456.m のようにしてメールで提出する．
あて先：`comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）：02-060123456
本文：もし今日の感想・意見・要望などがあれば書いてください．

研究

(1) 複素関数論で習った下記の複素数の値と右での *Mathematica* による出力結果（とその近似値）を比べよ：

$\sqrt{-1}$:	`Sqrt[-1]`
$(-1)^{1/3}$:	`(-1)^(1/3)`
$e^{\pi i/4}$:	`Exp[Pi*I/4]`
$\sin i$:	`Sin[I]`
$\arcsin 3$:	`ArcSin[3]`
$\log i$:	`Log[I]`
$\log(1+i)$:	`Log[1 + I]`
$(1-i)^i$:	`(1-I)^I`

(2) `FullSimplify` を用いて $a=\sqrt{\dfrac{\sqrt{6}+1}{\sqrt{6}-1}+\dfrac{\sqrt{6}+3}{\sqrt{6}+1}}$ を簡約化しても，2 重根号が残ってしまうことを確認せよ．次に $\sqrt{5}a$ を簡約化するとどうなるか？　**（97 名古屋学院大・改）**

(3) 次の問題を *Mathematica* が厳密値で解けるか試してみよ．近似値の場合はどうか？: $x=\sqrt[3]{7+5\sqrt{2}}$, $y=\sqrt[3]{7-5\sqrt{2}}$ のとき，$x+y=\boxed{}$, $xy=\boxed{}$ となるから，$x^4+y^4=\boxed{}$, $x^5+y^5=\boxed{}$ である．　**（84 早大・改）**
(Hint: $y^3=7-5\sqrt{2}<0$ であるから，$y<0$ としたいのだが…？)

(4) ドキュメントセンターをひらき，`Expand` や `Together` に関連してどのような組み込み関数があるか調べよ．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

リストと数列

作成日： October 20, 2011　Version： 1.2

Tips：評価の放棄

Mathematica に計算させていると，予想外に時間がかかってしまうことがある．入力式の評価を放棄 (abort) させるには，Command + .（ピリオド）を用いる.（Windows ではAlt + .となる． メニューから「評価」→「評価を放棄」を選んでもよい.）

リストとは？

リストとはベクトル，行列，テンソル，数列，集合などなど，さまざまなものを表現できる *Mathematica* 独自のデータ構造（データのいれもの）であり，中括弧 { } を用いて

`{a, b, c}`　`{{a, b}, {c, d}},`　`{Sin[x], Cos[x], Tan[x]}`

のように表される．

リストの計算

リストにはさまざまな演算が大胆に適用できる．

リスト v1 を定義：	`v1 = {a, b, c}`	各要素を 3 乗：	`v1^3`
リスト v2 を定義：	`v2 = {p, q, r}`	5 を「リスト乗」：	`5^v1`
ベクトル的な和：	`v1 + v2`	指数関数を各要素に作用：	`Exp[v1]`
ベクトル的な定数倍：	`100*v1`	リストの「リスト乗」：	`v1^v2`
後ろからの定数倍：	`v1*100`	要素ごとの積：	`v1*v2`
各要素に 1 を足す:	`v1 + 1`	要素ごとの商：	`v1/v2`
文字 x から各要素を引く：	`x - v1`	ベクトルとしての内積：	`v1.v2`

問題 1. 上の表のコマンドを入力し，どのような結果が得られるか確認せよ．

問題 2. `u = {1, 2, 3, 4, 5}`と定義する．これと上の表のコマンドを活用して，$\{x-1,\ (x-1)^2,\ (x-1)^3,\ (x-1)^4,\ (x-1)^5\}$,　$\left\{x,\ \frac{x^2}{2!},\ \frac{x^3}{3!},\ \frac{x^4}{4!},\ \frac{x^5}{5!}\right\}$ と等しいリストをそれぞれ作成せよ．ただし，n の階乗は `n!` もしくは `Factorial[n]` で計算できる．

リストの生成

リスト $\{1,2,3,4,5\}$ を定義：	`v1 = Range[5]`
リスト $\{4,5,\ldots,10\}$ を定義：	`v2 = Range[4,10]`
$4 \le x \le 10$ で 0.7 刻みのリスト：	`v3 = Range[4,10, 0.7]`
一般項でリストを作る（2 乗）：	`v4 = Table[n^2, {n,1,10}]`
（文字式）：	`v5 = Table[x^n, {n,0,10}]`
リストの結合：	`Join[v1, v2]`
リストの和集合：	`Union[v1, v2]`
リストの共通部分：	`Intersection[v1, v2]`

問題 3. 上の表のコマンドを入力し，どのような結果が得られるか確認せよ．

問題 4. `Table` を用いて，自然対数の底 e への収束列 $a_n = \left(1+\frac{1}{n}\right)^n$ の第 10 項目まで近似値（N 関数を施した値）からなるリスト `an` を作れ．

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

問題 5. Table を用いて，$x-1, x^2-1, \cdots, x^{10}-1$ からなるリストと，その各要素を因数分解した結果からなるリストを作れ．

リストのリストのリストの…

リストのリストを定義：	`m1 = {{a, b}, {c, d}, {f, g}}`
行列 (matrix) として表示：	`m1 //MatrixForm` もしくは `MatrixForm[m1]`
表 (table) として表示：	`m1 //TableForm` もしくは `TableForm[m1]`
一般項でリストを作る（和）：	`m2 = Table[i + j, {i,1,5}, {j,1,5}]`
（文字式）：	`m3 = Table[x^i*y^j, {i,1,5}, {j,1,5}]`

問題 6. 上の表のコマンドを入力し，どのような結果が得られるか確認せよ．

問題 7. (九九の表) リストの TableForm を用いて，1 から 9 までの積の表を作れ．また，1 から 21 までの奇数同士の積を表にせよ．

問題 8. (一覧表の作成) 「{n, n^2, n^3} の形のリスト」からなるリスト

```
pow = Table[{n, n^2, n^3}, {n,1,10}]
```

を作成せよ．さらに，これを MatrixForm および TableForm を用いて表せ．

リストから要素を取り出す

再度 2 乗 (square) の列：	`sq = Table[n^2, {n,1,10}]`
リスト sq の 7 番目の要素：	`sq[[7]]`
再度，和 (wa) の行列：	`wa = Table[i + j, {i,1,5}, {j,1,5}]`
再度行列として表示：	`wa //MatrixForm`
行列 wa の 2 行 3 列目：	`wa[[2,3]]`

問題 9. 上の表のコマンドを入力し，どのような結果が得られるか確認せよ．

問題 10. (収束列と誤差の一覧表)

(1) **問題 4** で作ったリスト an を用いて，a_n と e との誤差 $|a_n - e|$ からなるリスト anError を作れ．ただし，数 x の絶対値は関数 Abs[x] で与えられる（Hint: たとえば $|a_5 - e|$ はリスト an を使うと Abs[an[[5]]- E] で表される.）

(2) **問題 8** を応用して，a_n と誤差 $|a_n - e|$ を比較する一覧表を作成せよ．

数列の有限和と無限和

またまた 2 乗の列：	`sq = Table[n^2, {n, 1, 10}]`
1 から 10 までの 2 乗の和：	`s1 = Sum[n^2, {n, 1, 10}]`
一般項：	`s2 = Sum[k^2, {k, 1, n}]`
1 から 10 までの −2 乗の和：	`s3 = Sum[1/n^2, {n, 1, 10}]`
−2 乗の無限和（級数）：	`s4 = Sum[1/n^2, {n, 1, Infinity}]`
10 次多項式を作る：	`s5 = Sum[x^n/n!, {n, 0, 10}]`
じゃあこの級数は？:	`s6 = Sum[x^n/n!, {n, 0, Infinity}]`
数列の積は Product を使う：	`p = Product[2*n - 1, {k, 1, 10}]`
またまた和の行列：	`wa = Table[i + j, {i,1,5}, {j,1,5}]`
：	`wa //TableForm`
これらすべての和：	`Sum[i + j, {i,1,5}, {j,1,5}]`
一般項：	`Sum[i + j, {i,1,n}, {j,1,m}]`

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

問題 11. 上の表のコマンドを入力し，どのような結果が得られるか確認せよ．

問題 12.（収束速度の比較） `Table` と `Sum` を組み合わせて，e への収束列 $b_n = 1+\frac{1}{1!}+\frac{1}{2!}+\cdots+\frac{1}{n!}$ の第 10 項目までの近似値からなるリストを作成せよ．また，**問題 10** と同様にして，a_n，$|a_n - e|$，b_n，$|b_n - e|$ を比較する一覧表を作成せよ．

問題 13.（大学入試問題） *Mathematica* を用いて計算せよ．

(1) 次の和を求めよ：$\displaystyle\sum_{k=2}^{100} \frac{3}{\sqrt{k}+\sqrt{k-1}}$．（近似値だけでよい）　**（99 兵庫大）**

(2) 数列 $\frac{2}{2^2-1}, \frac{2}{3^2-1}, \frac{2}{4^2-1}, \ldots$ の第 n 項のまでの和を求めよ．　**（00 日本福祉大）**

(3) 連続する m 個の奇数 $1, 3, 5, \ldots, 2m-1$ の中から，異なる 2 つの数をとって積をつくる．こうして得られる ${}_m\mathrm{C}_2$ 通りの積すべての和を求めよ．　**（99 浜松医大）**

(4) 初項 4，公比 8 の等比数列の第 n 項までの和を S_n で表す．このとき，S_{99} は $\boxed{\text{カキ}}$ けたの整数，S_{100} は $\boxed{\text{クケ}}$ けたの整数である．さらに，S_1 から S_{100} の間で，$\boxed{\text{クケ}}$ 以下のすべてのけた数の S_n が存在するか確認せよ．また，S_n が初めて 150 けたを超えるのはいつかを，試行錯誤で確定せよ．　**（92 センター本試・改）**
(Hint: S_n の桁数は $1+\log_{10} S_n$ の整数部分で与えられる．その一覧表を作れ．)

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 03-060123456.m のようにしてメールで提出する．
あて先：`comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）：03-060123456
本文：もし今日の感想・意見・要望などがあれば書いてください．

おねがい：退出時，iMac をログアウトでなく**シャットダウン**してください．

研究

(1) i, j, k が 1 から n までの値をすべて動くとき，$i+j+k$ の和を求めよ．
すなわち，$\displaystyle\sum_{1\le i,j,k\le n} (i+j+k)$ を計算せよ．

(2) 一般項 $p_n = n^2+n+41$ で与えられる数列が初めて素数でなくなる n を求めよ．(ある m が素数かどうかを判定する組み込み関数は `PrimeQ[m]` である．)

(3)
$$s_n := \sum_{k=1}^{n} \cos\frac{2\pi k}{n}, \qquad p_n := \prod_{k=1}^{n} \cos\frac{2\pi k}{n}$$
とおく．$n = 5, 6, 7, 8, 9$ について s_n と p_n を計算せよ．また `FullSimplify` によって，s_n と p_n の一般項を予測せよ．(じつは，*Mathematica* は一般項を知っている．)

(4) ドキュメントセンターをひらき，リストに関連してどのような組み込み関数があるか調べよ．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

方程式とグラフ

作成日：October 27, 2011　Version：1.1

Tips：入力結果を省略させるには？長い式やリスト（数列や行列など）を `Table` などで定義すると，入力結果の表示だけで画面がいっぱいになってしまう．このような表示を省略したければ，入力式の最後にセミコロン ; をつけるとよい．

2 次方程式とグラフ

方程式は組み込み関数 `Solve` で解ける．答えは代入形（置換則リスト）で返ってくるので，扱いには注意が必要である．また解の様子をグラフで吟味するには，`Plot` を使う．

[1]	$x^2 - x - 1 = 0$ を解き，結果を `sol` とおく：	`sol = Solve[x^2 - x - 1 == 0, x]`
[2]	厳密解から近似値へ：	`N[sol]`
[3]	いきなり数値解（近似解）を求める：	`NSolve[x^2 - x - 1 == 0, x]`
[4]	有効数字を 100 桁に指定：	`NSolve[x^2 - x - 1 == 0, x, 100]`
[5]	グラフで確認（範囲は $-1 \le x \le 2$）：	`Plot[x^2 - x - 1, {x, -1, 2}]`

問題 1. 上の表のコマンドを入力し，結果を確認せよ．

問題 2. (解の図示) `Solve` 関数および `NSolve` 関数を用いて方程式 $x^3 - 19x + 30 = 0$ を解け．また $f(x) = x^3 - 19x + 30$ のグラフを描け．ただし，x 軸との交点が分かるように適当な描画範囲を選ぶこと．

方程式の解を活用する

[6]	代入（置換則）の復習．x^3 に $x = 5$ を代入：	`x^3 /. {x -> 5}`
[7]	さっきの方程式 $x^2 - x - 1 = 0$ の解：	`sol`
[8]	解だけからなるリストを作成：	`x /. sol`
[9]	解のリストに直接代入：	`{a, b} = x /. sol`
[10]	値を確認：	`a`
[11]	：	`b`
[12]	式 $t = x^3 - 1/x^3$ の値を求めよう：	`t = x^3 - 1/x^3`
[13]	`t` に解を両方とも（`a` と `b`）代入：	`ans = t /. sol`
[14]	値を整理：	`Simplify[ans]`
[15]	`t` に `sol` のひとつ目の解（`a` だけ）を代入：	`ans1 = t /. sol[[1]]`
[16]	ふたつ目の解（`b` だけ）を代入：	`ans2 = t /. sol[[2]]`

問題 3. 上の表のコマンドを入力し，結果を確認せよ．

問題 4. 方程式 $x^2 - x - 5 = 0$ の解を α, β $(\alpha < \beta)$ とするとき，$\alpha^4 + \beta^4$ および $\alpha^4 - \alpha^2 - 30$ を計算せよ．([9]-[11] を使う．)

問題 5. (大学入試問題) 結果には必要に応じて `Simplify` もしくは `FullSimplify` を施すこと．

(1) $x^2 - 3x + 1 = 0$ のとき，$x^3 + \dfrac{1}{x^3}$, $x^5 + \dfrac{1}{x^5}$ の値を求めよ．　**(97 帝塚山大)**

(2) $x^2 + \dfrac{1}{x^2} = 4$ のとき，次の値を求めよ．ただし，$1 < x < 2$ とする．

（ア）$x^2 - \dfrac{1}{x^2}$　（イ）$x^3 - \dfrac{1}{x^3}$　（ウ）$x^5 - \dfrac{1}{x^5}$　**(01 大同工大)**

(Hint: `sol = Solve[...]` で解いたあと，`sol` に `N` 関数を施す．$1 < x < 2$ となる解が `i` 番目であれば，(ア)〜（ウ）に `sol[[i]]` を代入．)

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

連立方程式

[17]	直線 vs 直線：	`s1 = Solve[{x + y == 1, x - y == -2}, {x, y}]`
[18]	$y =$ の形のグラフで見る：	`g1 = Plot[{-x + 1, x + 2}, {x, -5, 5}]`
[19]	陰関数でグラフを描く： （範囲は $-5 \leq x, y \leq 5$）：	`g12 = ContourPlot[{x + y == 1, x - y == -2},` `{x, -5, 5}, {y, -5, 5}]`
[20]	直線 vs 放物線：	`s2 = Solve[{x + y == 1, y == x^2 - 1}, {x, y}]`
[21]	陰関数のグラフ： （範囲は $-4 \leq x, y \leq 4$）：	`g2 = ContourPlot[{x + y == 1, y == x^2 - 1},` `{x, -4, 4}, {y, -4, 4}]`

問題 6. 上の表のコマンドを入力し，結果を確認せよ．

問題 7. (直線 vs 円) [20][21] を応用して，連立方程式 $x + y = 1$, $x^2 + y^2 = 2$ を解き，陰関数としてのグラフを $-3 \leq x, y \leq 3$ の範囲で描け．

問題 8. (連立三角方程式) 同様にして連立方程式 $\sin x + \sin y = \dfrac{\sqrt{3}}{2}$, $\cos x + \cos y = \dfrac{1}{2}$ を解き．陰関数としてのグラフを $-2\pi \leq x, y \leq 2\pi$ の範囲で描け． **（63 滋賀大・改）**

連立方程式の解の活用

[22]	さっきの連立方程式（直線 vs 放物線）の解：	`s = s2`
[23]	解をベクトル v1 と v2 に代入：	`{v1, v2} = {x, y} /. s`
[24]	値を確認：	`v1`
[25]	：	`v2`
[26]	解を用いた計算：	`h = x^5 + y^5;`
[27]	両方の解（v1 と v2）を代入：	`ans = h /. s`
[28]	最初の解（v1）だけ代入：	`ans = h /. s[[1]]`

問題 9. 上の表のコマンドを入力し，どのような結果が得られるか確認せよ．

問題 10. (3 元連立方程式) **（明治薬大・改）**

(1) x, y, z が方程式 $x+y+z = 2$, $x^2+y^2+z^2 = 6$, $x^3+y^3+z^3 = 8$ を満たすとき，$xy+yz+zx$ および xyz の値をもとめよ．さらに，$x^2 - y + z$ が最大になるとき，x, y, z の値を求めよ．

(2) 下のコマンドを入力して 3 つの方程式の陰関数グラフ（曲面）を描き，これらの曲面の共通交点が 6 つあることを確かめよ．：

```
gr = ContourPlot3D[
        {x + y + z == 2, x^2 + y^2 + z^2 == 6, x^3 + y^3 + z^3 == 8},
        {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
        ContourStyle -> {Yellow, Green, Orange},
        Mesh -> None, AxesLabel -> Automatic ]
```

課題の提出： 以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 04-060123456.m のようにしてメールで提出する．
あて先： comp0-2011@math.nagoya-u.ac.jp
件名（半角で！）： 04-060123456
本文： もし今日の感想・意見・要望などがあれば書いてください．

おねがい： 退出時，iMac をログアウトでなく**シャットダウン**してください．

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

微分・積分・いろんなグラフ

作成日： November 10, 2011 Version： 1.2

Tips： ノートブックはセル（]）で分割・グループ化されている．タイトル・小見出しなどのセルを作るには，以下のショートカットがおすすめ：

でかい字のタイトル (title)：	Command + 1
節 (section)，横線が入る：	Command + 4
小見出し ((sub)subsection)：	Command + 5 もしくは 6

◎今日の課題◎：問題の前にある [1][2]…等のコマンドをまず入力し，結果を確かめた上で問題にとりかかること．

関数の定義（即時割り当て，自分で関数を作る）

[1]	関数の定義（即時割り当て）：	`f[x_] = x^2 + 1`
[2]	複素数 $x = 1 + i$ を代入：	`f[1 + I]`
[3]	リスト {2,3} を代入：	`f[{2, 3}]`
[4]	2 変数関数：	`g[x_, y_] = x y + 1`
[5]	値を代入：	`g[3, 3]`

1 変数関数の微分

[6]	関数 f を再定義：	`f[x_] = x^3 - 3 x;`
[7]	x で微分（文字式 d1 とおく）：	`d1 = D[f[x], x]`
[8]	$x = 2$ での微分の値：	`d1 /. {x -> 2}`
[9]	2 階微分：	`d2 = D[f[x], {x, 2}]`
[10]	こららの関数のグラフを描く：	`Plot[{f[x], d1, d2}, {x, -4, 4}]`
[11]	f の導関数を定義：	`g[x_] = D[f[x], x]`
[12]	$x = 2$ での導関数の値：	`g[2]`

問題 1. (極値を求める) 上の [11] で定義した `f[x]` の導関数 `g[x]` が 0 となる `x` を `Solve` 関数を用いて求めよ．さらに，そのときの関数 `f[x]` の値を求めよ．(Hint: 前回のプリント [6]-[13] を参照．)

問題 2. (n 階導関数の一覧)

(1) `Table` 関数をもちいて，$f(x) = x^2 \sin x$ の 0 階から 5 階微分までからなるリスト `deriv` を作成せよ．また，`TableForm` を適用し一覧表をつくれ．ただし，関数 `f[x]` の n 階導関数は `D[f[x], {x, n}]` で計算できる．

(2) [10] をまねて，`deriv` の各要素のグラフをまとめて描け．(描画範囲は各自のセンスで決めてよい．)

多変数関数の偏微分

[13]	関数 f を再定義：	`f[x_, y_] = x^2 + y^2`
[14]	f のグラフを描く：	`r = 3; Plot3D[f[x, y], {x, -r, r}, {y, -r, r}]`
[15]	x もしくは y で偏微分：	`d1 = D[f[x, y], x]`
[16]	：	`d2 = D[f[x, y], y]`

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

[17]	2 階の偏微分，$\partial^2/\partial^2 x$：	`d11 = D[f[x, y], x, x]`
[18]	$\partial^2/\partial x\partial y$：	`d12 = D[f[x, y], x, y]`
[19]	$\partial^2/\partial^2 y$：	`d22 = D[f[x, y], y, y]`
[20]	f の x 偏導関数 g を定義：	`g[x_,y_] = D[f[x, y], x]`
[21]	f と g のグラフを描く：	`Plot3D[{f[x, y], g[x, y]}, {x, -r, r},`
	（色をつける描画オプション→）：	`{y, -r, r}, PlotStyle -> {Blue, Green}]`

問題 3. (ラプラシアン) [17][19] を応用して，$f(x,y)=e^x \sin y$ は $\left(\dfrac{\partial^2}{\partial x^2}+\dfrac{\partial^2}{\partial y^2}\right)f(x,y)=0$ を満たすことを示せ．また，$g(x,y,z)=\dfrac{1}{\sqrt{x^2+y^2+z^2}}$ は $\left(\dfrac{\partial^2}{\partial x^2}+\dfrac{\partial^2}{\partial y^2}+\dfrac{\partial^2}{\partial z^2}\right)g(x,y,z)=0$ を満たすことを示せ．

問題 4. (極値をもとめる) 関数 $f(x,y)=x^3-y^3-3x+12y$ を考える．

(1) 偏導関数 f_x, f_y をそれぞれ `fx`，`fy` として定義せよ．さらに `Solve` 関数を用いて，極値を与える点の候補 (`sol`) を求めよ．また，そこでの f の値を求めよ．

(2) $f_{xx}f_{yy}-f_{xy}^2$ で定まる関数を `g` とおき，`sol` を代入してその正負を確認せよ．

(3) `Plot3D` を用いて，$f(x,y)$ のグラフを描け．ただし，適宜

`PlotRange -> {-30, 30}, AxesLabel -> {x, y}`

などのオプション描画オプションをつけて，極値がどこかわかるように描くこと．

積分

[22]	**不定積分** $\int \sin x\,dx$：	`int = Integrate[Sin[x], x]`
[23]	グラフを描く：	`Plot[{Sin[x], int}, {x, 0, Pi}]`
[24]	**定積分** $\int_0^2 (x^2-2)dx$：	`Integrate[x^2-2, {x, 0, 2}]`
[25]	積分される部分を図示：	`Plot[x^2-2, {x, 0, 2}, Filling -> Axis]`
[26]	**広義積分** $\int_0^\infty \dfrac{\sin x}{x}dx$ の値：	`Integrate[Sin[x]/x, {x, 0, Infinity}]`
[27]	積分される部分を図示：	`Plot[Sin[x]/x, {x, 0, 10 Pi},`
	（描画オプションを指示→）：	`PlotRange -> {-0.5, 1.0}, Filling -> Axis]`
[28]	関数 $g(x)=\int_0^x \dfrac{\sin t}{t}dt$ を定義：	`g[x_] = Integrate[Sin[t]/t, {t, 0, x}];`
[29]	グラフにしてみる：	`Plot[{Sin[x]/x, g[x]}, {x, 0, 15 Pi}]`

問題 5. (x の x 乗) $x>0$ にたいし，$f(x)=x^x$，$g(x)=f'(x)$，$h(x)=\int_0^x f(t)dt$ とさだめる．これらのグラフを $0\le x\le 2$ の範囲で描け．(h の計算にはやや時間がかかる.)

3 次元グラフいろいろ

[30]	関数 f を再定義：	`f[x_, y_] = (x^2 + 3 y^2) Exp[1 - x^2 - y^2]`
[31]	3 次元のグラフ：	`Plot3D[ f[x, y], {x, -2, 2}, {y, -2, 2}]`
[32]	等高線グラフ：	`ContourPlot[ f[x, y], {x, -2, 2}, {y, -2, 2}]`
[33]	等高線＝陰関数グラフ：	`ContourPlot[ f[x, y] == 1, {x, -2, 2}, {y, -2, 2}]`
[34]	密度プロット：	`DensityPlot[ f[x, y], {x, -2, 2}, {y, -2, 2}]`

問題 6. **問題 4** の $f(x,y)$ について，`ContourPlot` および `DensityPlot` によるグラフを描け．`ContourPlot` においては描画範囲やオプション `Contours -> 30` などを加え，$f(x,y)$ の極大・極小点および鞍点がわかるように工夫せよ．

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 05-060123456.m のようにしてメールで提出する．
あて先：`comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）：05-060123456
本文：もし今日の感想・意見・要望などがあれば書いてください．

おねがい：退出時，iMac をログアウトでなく**シャットダウン**してください．

研究・ちからだめし

問題 7. **(大学入試問題)**

(1) 2 次関数 $f(x)$ が $f(2)=26$, $f'(2)=18$, $\int_{-1}^{1} f(x)dx = 6$ をみたすとき，$f(x)$ をもとめよ．
（Hint: $f(x)=ax^2+bx+c$ とおいて a,b,c について `Solve`）　**(03 近畿大)**

(2) 整式 $f(x)$ と実数 C が $\int_0^x f(y)dy + \int_0^1 (x+y)^2 f(y)dy = x^2 + C$ をみたすとき，この $f(x)$ と C を求めよ．
（Hint: 等式をよく眺めると，$f(x)$ は 2 次以下であることがわかる．）　**(09 京大)**

(3) x の 3 次関数 $f(x)=ax^3+bx^2+cx+d$ が 3 つの条件 $f(1)=1$, $f(-1)=-1$, $\int_{-1}^{1}(bx^2+cx+d)=1$ をすべて満たしているとする．このような $f(x)$ の中で定積分 $I=\int_{-1}^{\frac{1}{2}} f''(x)^2 dx$ を最小にするものを求め，そのときの I の値を求めよ．（Hint: まず 3 条件から b,c,d を a の式として `Solve`．このとき，I は a の関数となる．）　**(11 東京大)**

(4) 2 次関数 $f(x)$ と定数 p が $\int_0^x f(t)dt + x\int_{-1}^{1} f(t)dt - \frac{1}{3}\{f(1)-f(-1)\} = 4x^3+px^2-10x-4$ を満たす．

(a) 2 次関数 $f(x)$ と定数 p の値を求めよ．
(b) $g(x)=(x+1)f(x)$ とする．関数 $g(x)$ の極値とそのときの x の値をもとめよ．

(07 同志社大)

問題 8. 恒等式をもとめる関数 `SolveAlways` というのがある．ドキュメントセンターで使い方をしらべて，上の (2)，(4) を解くのに応用せよ．（もちろん，使わなくても工夫次第で解ける．）

問題 9. 関数 $h:\mathbb{R}^2 \to \mathbb{R}$ を $(x,y)\neq(0,0)$ のとき $h(x,y)=\dfrac{xy(x^2-y^2)}{x^2+y^2}$，$h(0,0)=0$ によって定める．この関数は原点で $h_{xy}\neq h_{yx}$ となる有名な例だが，*Mathematica* で計算させると $h_{xy}=h_{yx}$ となることを確認せよ．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

行列と１次変換・ParametricPlot

作成日：November 17, 2011　Version：1.2

Tips　：　グラフのスタイルは**「オプション」**で調整する．たとえば Plot の場合，

[a]	Plot のデフォルト値：	`Plot[Sin[x], {x, -Pi, Pi}]`
[b]	縦横の縮尺をあわせる：	`Plot[（同上）, AspectRatio -> Automatic]`
[c]	x 軸と囲まれる部分を塗る：	`Plot[（同上）, Filling -> Axis]`
[d]	グラフの線を緑色で太く：	`Plot[（同上）, PlotStyle -> {Green, Thick}]`

オプションはたくさんある（`Options[Plot]` で一覧が見れる）．使いこなすには，ドキュメントセンター（ヘルプ）を参照しよう．

◎！注意！◎：問題だけを解くのではなく，[1][2]…等のコマンドも必ず入力し，結果を確かめること．ファイルの提出時もこれらの入力内容は消さずに残しておくこと．

行列とベクトルの演算

[1]	ベクトルを定義：	`v1 = {x, y}; v2 = {z, w};`
	行列を定義：	`m1 = {{a, b}, {c, d}};`
	：	`m2 = {{3, 2}, {1, 4}};`
[2]	行列として表示：	`m1 //MatrixForm`　もしくは `MatrixForm[m1]`
[3]	ベクトルの内積：	`v1.v2`
[4]	ベクトルと行列の積：	`m1.v1`
[5]	行列と行列の積：	`pr = m1.m2;`
	：	`pr // MatrixForm`
[6]	行列の**べき乗**（7 乗）：	`MatrixPower[m2, 7] // MatrixForm`
[7]	行列の**転置**：	`Transpose[m1] // MatrixForm`
[8]	**逆行列**：	`Inverse[m1] // MatrixForm`
[9]	**行列式**：	`Det[m1]`
[10]	**トレース**：	`Tr[m1]`
[11]	**単位行列**・2 次：	`i2 = IdentityMatrix[2]`
[12]	3 次：	`i3 = IdentityMatrix[3]`

問題 1. (ケーリー・ハミルトンの等式) [5][9][10][11] をもちいて，任意の 2 次正方行列 $M = \begin{pmatrix} a & b \\ c & d \end{pmatrix}$ は等式 $M^2 - \mathrm{tr}(M)M + \det(M)I_2 = O$ をみたすことを示せ．ただし I_2 は 2 次の単位行列．

固有値・固有ベクトル

[13]	行列を定義：	`m = {{3, 2}, {1, 4}};`
[14]	**固有値・固有ベクトル**を求める：	`es = Eigensystem[m]`
[15]	対応する固有値・ベクトルは縦に並ぶ：	`es // MatrixForm`
[16]	要素をとりだして…：	`lam = es[[1, 1]]; v = es[[2, 1]];`
[17]	定義に当てはめる：	`m.v == lam v`
[18]	固有多項式：	`p[x_] = Det[m - x i2]`
[19]	固有値の確認：	`Solve[p[x] == 0, x]`

問題 2. (対角化) 上の `es` をもちいて，`q = Transpose[es[[2]]]` とおくとこれが `m` の対角化行列を与える．その理由を考えた上で，`Inverse[q].m.q` を計算し，確かに対角化されていることを確認せよ．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

1 次変換（2 次元）と ParametricPlot

1 次変換 $f:\mathbb{R}^2\to\mathbb{R}^2$ を $f\begin{pmatrix}x\\y\end{pmatrix}=\begin{pmatrix}3&2\\1&4\end{pmatrix}\begin{pmatrix}x\\y\end{pmatrix}$ で定義する．その作用を可視化しよう．

[20]	さっきの行列を確認：	`m // MatrixForm`
[21]	1 次変換を関数として定義：	`f[{x_, y_}] = m.{x, y}`
[22]	具体的な値で確認：	`v = {-5, 6}; f[v]`
[23]	円のパラメーター表示：	`ParametricPlot[{Cos[t], Sin[t]}, {t, 0, 2 Pi}]`
[24]	f による像：	`ParametricPlot[f[{Cos[t], Sin[t]}], {t, 0, 2 Pi}]`
[25]	円を表すベクトル値関数：	`circ[t_] = {Cos[t], Sin[t]};`
[26]	円とその像を比較：	`ParametricPlot[{circ[t], f[circ[t]]}, {t, 0, 2 Pi}]`
[27]	領域 $-1\le x,y\le 1$：	`ParametricPlot[{s, t}, {s, -1, 1}, {t, -1, 1}]`
[28]	その像：	`ParametricPlot[f[{s, t}], {s, -1, 1}, {t, -1, 1}]`
[29]	比較：	`ParametricPlot[{{s, t}, f[{s, t}]},`
	：	`{s, -1, 1}, {t, -1, 1}]`

問題 3.（リサジュー曲線） 整数 m, n を適当に選び，$\begin{pmatrix}\cos mt\\\sin nt\end{pmatrix}$ $(0\le t\le 2\pi)$ で得られる曲線のグラフ ParametricPlot で描け．(整数をいろいろ試し，一番「美しい」と思えるグラフをみつけよ.）また，その f による像のグラフも描け．([25][26] を参考にせよ.）

問題 4.（単位円板の像） 関数 $disk(r,t)=(r\cos t, r\sin t)$ （ただし $0\le r\le 1, 0\le t\le 2\pi$）を定義し，単位円板とその f による像を ParametricPlot で図示せよ．([29] を参考にせよ.）

1 次変換（3 次元）と ParametricPlot3D

1 次変換 $g:\mathbb{R}^3\to\mathbb{R}^3$ を $g\begin{pmatrix}x\\y\\z\end{pmatrix}=\begin{pmatrix}1&0&2\\0&2&0\\2&0&1\end{pmatrix}\begin{pmatrix}x\\y\\z\end{pmatrix}$ で定義する．その作用を可視化しよう．

[30]	行列を定義：	`mm = {{1, 0, 2}, {0, 2, 0}, {2, 0, 1}};`
[31]	：	`mm // MatrixForm`
[32]	固有値・ベクトルを確認：	`Eigensystem[mm] // MatrixForm`
[33]	1 次変換を定義：	`g[{x_, y_, z_}] = mm.{x, y, z}`
[34]	螺旋 (spiral) を定義：	`spi[t_] = {Cos[t], Sin[t], t/10}`
[35]	その 3 次元グラフ：	`ParametricPlot3D[spi[t], {t, 0, 10 Pi}]`
[36]	g による像：	`ParametricPlot3D[g[spi[t]], {t, 0, 10 Pi}]`
[37]	比較する：	`ParametricPlot3D[{spi[t], g[spi[t]]}, {t, 0, 10 Pi}]`
[38]	球面（sphere）を定義：	`sph[s_, t_] = {Cos[s] Sin[t], Sin[s] Sin[t], Cos[t]};`
[39]	3 次元グラフとその像：	`ParametricPlot3D[`
	：	`{sph[s, t], g[sph[s, t]]}, {s, 0, 2 Pi}, {t, 0, Pi},`
	描画オプションを指定→：	`PlotStyle -> {Opacity[0.9, Green], Opacity[0.4, Yellow]}]`

問題 5.（トーラス） [38][39] を応用し，以下で定義されるトーラス (torus，円環面) について，そのグラフ，その像のグラフを描け．また，それらを同一空間に描画せよ．

```
tor[s_, t_] = {Cos[s] (3 + Cos[t]), Sin[s] (3 + Cos[t]), Sin[t]}
```

ただし，s，t の範囲は 0 から 2π までとする．

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

課題の提出： 以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 06-060123456.m のようにしてメールで提出する．
あて先： comp0-2011@math.nagoya-u.ac.jp
件名（半角で！）： 06-060123456
本文： もし今日の感想・意見・要望などがあれば書いてください．

おねがい： 退出時，iMac をログアウトでなく**シャットダウン**してください．

研究

問題 6. (よりよい可視化) [38][39] の絵は美しいが，1 次変換の作用がわかりづらい．(たとえば，この絵からは向きを保つかどうかもわからない.) そこで，ちょっとした工夫をしてみよう．以下を実行し，1 次変換の作用を確認せよ：

[40] 3 枚の板を定義： `xy[s_, t_] = {s, t, 0};`
[41] ： `yz[s_, t_] = {0, s, t};`
[42] ： `zx[s_, t_] = {t, 0, s};`
[43] 板の絵： `ParametricPlot3D[{xy[s, t], yz[s, t], zx[s, t]},`
： `{t, -1, 3}, {s, -1, 3}, PlotStyle -> {Red, Green, Blue}]`
[44] 板の g による像： `ParametricPlot3D[{g[xy[s, t]], g[yz[s, t]], g[zx[s, t]]},`
： `{t, -1, 3}, {s, -1, 3}, PlotStyle -> {Red, Green, Blue}]`

最後に，[43][44] の絵を同一空間に描け．

問題 7. 2 次元の一次変換 f についても，何らかの工夫をして作用を可視化せよ．

問題 8. 一次変換 f に一見よく似た関数 $S : \mathbb{R}^2 \to \mathbb{R}^2$ を $S\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} \sin(3x+2y) \\ \sin(x+4y) \end{pmatrix}$ で定義する．この作用を問題 4 の円板（$disk$）関数を用いて可視化せよ．また，半径 $r \le 0.2$ 程度の範囲であれば，その絵は f による像とほとんど変わらない．その理由を考えよ．

問題 9. (土星はかっこいい！) 土星（っぽいもの）を描け．

問題 10. (大学入試問題) xy 平面上で t を変数とする媒介変数表示 $\begin{cases} x = 2t + t^2 \\ y = t + 2t^2 \end{cases}$ で表される曲線を C とする．

(1) $t \neq -1$ のとき，$\dfrac{dy}{dx}$ を t の式で表せ．

(2) 曲線 C 上で $\dfrac{dy}{dx} = -\dfrac{1}{2}$ を満たす点 A の座標を求めよ．

(3) 曲線 C 上の点 (x, y) を点 (X, Y) に移す移動が $\begin{cases} X = \dfrac{1}{\sqrt{5}}(2x - y) \\ Y = \dfrac{1}{\sqrt{5}}(x + 2y) \end{cases}$ で表されているとする．このとき，Y を X を用いて表せ．

(4) 曲線 C の概形を xy 平面上にかけ． **（東京大 06）**

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

テイラー・フーリエ・Manipulate

作成日： November 24, 2011 Version : 1.2

Tips： Manipulate 関数の実行結果の中では「Dynamic 変数」というものが使われており，ノートブック中の他の計算と互いに影響しあっている．悪影響を避けるためには，

- Manipulate の実行結果は，確認したらすぐに消す（Delete）．
- 変数は頻繁に Clear する．もしくは，できるだけ別の文字を用いるようにする．

◎！注意！◎：問題だけを解くのではなく，[1][2]…**等のコマンドも必ず入力し，結果を確かめること．ファイルの提出時もこれらの入力内容は消さずに残しておくこと．**

関数の定義（即時割り当て vs. 遅延割り当て）

関数を定義する際には，**即時割り当て**と**遅延割り当て**の二種類があり，うまく使い分ける必要がある．その違いを，0 と 1 の間の実数をランダムにとる組み込み関数 RandomReal[] を用いて確認してみよう．

[1]	0 と 1 の間のランダムな実数：	`RandomReal[]`
[2]	:= で**遅延割り当て**：	`g[x_] := x * RandomReal[]`
[3]	毎回定義どおり評価される：	`{g[10], g[10], g[10]}`
[4]	一方，**即時割り当て**は一旦評価され…：	`h[x_] = x * RandomReal[]`
[5]	あとは毎回同じ結果となる：	`{h[10], h[10], h[10]}`

問題 1.（微分と遅延割り当て） `f[x_] = Sin[x], g[x_]:= D[f[x], x], h[x_]= D[f[x], x]` と定義する．このとき，`g[x]` は `Plot` でグラフが描けないが，`h[x]` はグラフが描けることを確認せよ．また，`Evaluate[g[x]]` とすれば描けるようになることを確認せよ．

マニピュレート（Manipulate）：動く式

[6]	因数分解の Table：	`Table[Factor[x^n - 1], {n, 1, 10}]`
[7]	Table を Manipulate に変えてみると…：	`Manipulate[Factor[x^n - 1], {n, 1, 10}]`
[8]	n を整数に限定させる：	`Manipulate[Factor[x^n - 1], {n, 1, 10, 1}]`

問題 2.（行列のべき乗） Manipulate を用いて行列 $M = \begin{pmatrix} 3 & 2 \\ 1 & 4 \end{pmatrix}$ のべき乗 M^p を $p = 1$ から 10 まで整数で変化させよ.（ちゃんと MatrixForm で見やすくすること.）

マニピュレート（Manipulate）：動くグラフ

[9]	グラフの Table：	`Table[Plot[Sin[k x], {x, -Pi, Pi}], {k, -5, 5}]`
[10]	Table を Manipulate に：	`Manipulate[Plot[Sin[k x], {x, -Pi, Pi}], {k, -5, 5}]`
[11]	描画範囲を固定する：	`Manipulate[Plot[(同上), PlotRange -> {-1, 1}]`
	：	`,{k, -5, 5}]`
[12]	k を整数に制限：	`Manipulate[ (同上) , {k, -5, 5, 1}]`
[13]	動く接線を描くぞー！：	`Clear[f, g, x]`
[14]	関数・導関数を定義：	`f[x_] = x^3 - 3 x; g[x_] = D[f[x], x];`
[15]	**接線** (tangent line) を定義：	`tanLine[x_, p_] = g[p] (x - p) + f[p]`
[16]	接線を動かす：	`Manipulate[`
	：	`Plot[{f[x], tanLine[x, p]}, {x, -3, 3},`
	：	`PlotRange -> {-5, 5}], {p, -3, 3}]`

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

問題 3.（ガタガタ） [16] で `PlotRange -> {-5, 5}]` を外すと，グラフがガタガタして情けない感じとなることを確認せよ．

問題 4.（一様収束？） [12] を応用して，$f_k(x) = x^k$ $(k = 1, 2, \ldots)$ がどこで一様収束しているかがわかる「動くグラフ」をつくれ．

問題 5.（接線） [14][15][16] を応用して，好きな関数の「動く接線グラフ」をつくれ．(他人に見せてそこそこ面白いものがよい.)

テイラー展開

[17]	関数を定義：	`f[x_] = Sin[x];`
[18]	x=0 での 5 次テイラー展開：	`Series[f[x], {x, 0, 5}]`
[19]	`x = Pi/4` での展開：	`Series[f[x], {x, Pi/4, 5}]`
[20]	関数部分を取り出す：	`Normal[Series[f[x], {x, 0, 5}]]`
[21]	テイラー多項式を遅延的に定義：	`g[x_, a_, n_]:= Normal[Series[f[x], {x, a, n}]]`
[22]	即時割り当てだと失敗する：	`h[x_, a_, n_] = Normal[Series[f[x], {x, a, n}]]`
[23]	`Evaluate` をつけて `Plot`：	`Plot[Evaluate[g[x, 0, 5]], {x, - Pi, Pi}]`
[24]	`Manipulate` で n を変化させる：	`Manipulate[Plot[Evaluate[{g[x, 0, n], f[x]}]`
	：	`,{x, -4 Pi, 4 Pi}, PlotRange -> {-3, 3}]`
	：	`,{n, 1, 30, 1}]`
[25]	a も変化させる：	`Manipulate[Plot[Evaluate[{g[x, a, n], f[x]}]`
	：	`,{x, -4 Pi, 4 Pi}, PlotRange -> {-3, 3}]`
	：	`,{n, 1, 30, 1}, {a, -2 Pi, 2 Pi}]`

問題 6.（テイラー展開） [25] を応用して，好きな関数（もちろん `Sin` 以外の，なんか面白い関数）の「動くテイラー多項式近似」をつくれ．

フーリエ展開

区分的に連続な関数 $f : \mathbb{R} \to \mathbb{R}$ および $n = 0, 1, 2, \ldots$ にたいし，

$$a_n = \frac{1}{\pi}\int_0^{2\pi} f(x)\cos nx\,dx, \quad b_n = \frac{1}{\pi}\int_0^{2\pi} f(x)\sin nx\,dx,$$

と定義する．このとき，

$$f(x) \sim \frac{a_0}{2} + \sum_{n=1}^{\infty}\{a_n \cos nx + b_n \sin nx\} \tag{1.1}$$

と表し，右辺を $f(x)$ の**フーリエ展開** (Fourier expansion) と呼ぶ．

[26]	関数の定義：	`f[x_] = x`
[27]	関数 `f[x]` のフーリエ展開：	`FourierTrigSeries[f[x], x, 10]`
[28]	関数として遅延的に定義：	`g[x_, n_] := FourierTrigSeries[f[x], x, n]`
[29]	`Manipulate` で近似の様子を確認：	`Manipulate[Plot[Evaluate[g[x, n]],`
	：	`{x, -2 Pi, 2 Pi}, PlotRange -> {-4, 4}]`
	：	`, {n, 1, 100, 1}]`

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

問題 7. [29] を応用して，$f(x) = x^2$ について「動くフーリエ展開」をつくれ．ただし，適当に `PlotRange` を変えて，大局的な様子がわかるようにすること．

問題 8. (矩形波，Square wave のフーリエ展開) [28] の代わりに次の関数でおきかえて，[29] のように収束の様子を視覚化せよ．（HINT: `Sum` を用いる．）

$$g_n(x) := \frac{4}{\pi}\left(\sin x + \frac{\sin 3x}{3} + \frac{\sin 5x}{5} + \cdots + \frac{\sin(2n-1)x}{(2n-1)}\right)$$

定理（フーリエ展開の収束）． 関数 $f : \mathbb{R} \to \mathbb{R}$ は次の性質を満たすとする：

- 周期 2π の関数である．すなわち，$f(x) = f(x + 2\pi)$.
- 区間 $-\pi < x < \pi$ では $f'(x)$ が存在し，連続.
- 極限 $f(-\pi + 0) := \lim_{x \to -\pi+0} f(x)$ および $f(\pi - 0) := \lim_{x \to \pi-0} f(x)$ が存在する.

このとき，連続点 $x \neq (2n+1)\pi$ $(n \in \mathbb{Z})$ において

$$f(x) = \frac{a_0}{2} + \sum_{n=1}^{\infty} \{a_n \cos nx + b_n \sin nx\}$$

であり，不連続点 $x = (2n+1)\pi$ $(n \in \mathbb{Z})$ においては

$$\frac{f(-\pi+0) + f(\pi-0)}{2} = \frac{a_0}{2} + \sum_{n=1}^{\infty} \{a_n \cos nx + b_n \sin nx\}$$

が成り立つ.（左右からの極限値の平均！）

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 07-060123456.m のようにしてメールで提出する．
あて先：comp0-2011@math.nagoya-u.ac.jp
件名（半角で！）：07-060123456
本文：もし今日の感想・意見・要望などがあれば書いてください．

おねがい：退出時，iMac をログアウトでなく**シャットダウン**してください．

研究

問題 9. (リサジュー曲線) `Manipulate` と `ParamericPlot` をもちいて，リサジュー曲線 $\begin{pmatrix} \cos mt \\ \sin nt \end{pmatrix}$ が n, m に応じて変化する様子を確認せよ．

問題 10. (等位面の変化) `Manipulate` と `ContourPlot3D` をもちいて，陰関数 $xyz = k$ $(k \in \mathbb{R})$ で表される 3 次元空間内の曲面が k に応じて変化する様子を確認せよ．

問題 11. (ドーナツを食べる) トーラスを

```
tor[u_, v_] = {Cos[u] (3 + Cos[v]), Sin[u] (3 + Cos[v]), Sin[v]};
```

で定義する．以下を実行し，トーラスの断面を確認せよ．

```
Manipulate[
    ParametricPlot3D[tor[u, v], {u, 0, 2 Pi}, {v, 0, 2 Pi},
        PlotRange -> {{-4, 4}, {a, 4}, {-1, 1}}, PlotStyle-> Brown]
            ,{a, -4, 4}]
```

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

複素級数・複素関数の可視化

作成日： December 01, 2011　Version： 1.1

◎！注意！◎：問題だけを解くのではなく，[1][2]…等のコマンドも必ず入力し，結果を確かめること．ファイルの提出時もこれらの入力内容は消さずに残しておくこと．

今回のテーマは「複素数を視る」ことである．すでに複素関数論を学んだ受講者の皆さんには，それなりの「鑑識眼」があるはず．描くことよりもぜひ，鑑賞するほうに力を注いでほしい．

必要な関数の準備

[1]	複素数から（実部，虚部）のベクトル (vector) へ：	`vect[z_] := {Re[z], Im[z]}`
[2]	たとえば…：	`vect[3 + 4 I]`
[3]	（実部，虚部）のベクトルから複素数 (complex) へ：	`comp[{x_, y_}] := x + y I`
[4]	たとえば…：	`comp[{3, 4}]`

方程式の虚数解をプロットする

[5]	方程式の左辺を定義：	`f[z_] = z^11 - 1;`
[6]	方程式を数値的に解く：	`sol = NSolve[f[z] == 0, z]`
[7]	ベクトルに直す：	`v = vect[z] /. sol`
[8]	平面上に図示：	`ListPlot[v]`
[9]	リストの順番に線で結ぶ：	`ListLinePlot[v, PlotMarkers -> Automatic]`

問題 1.（より見やすく） [8][9] にオプション `AspectRatio -> Automatic` を加えて，タテヨコの比率をあわせよ．また，[9] のオプション `PlotMarkers -> Automatic` を消したらどうなるか確認せよ．

問題 2. [5] の方程式を他の高次方程式に変えて，その解を [8][9] のように図示せよ．

`Manipulate` で `Locator` を使う

`Locator` はマウスの動きをそのまま `Manipulate` のパラメーターとして取り出すことができる直感的インターフェイスである．

[10]	`{x,y}` 方向の半直線グラフ：	`ray[{x_, y_}]:= ParametricPlot[`
	：	`t {x, y}, {t, 0, 5}, PlotRange -> 5]`
[11]	たとえば…：	`ray[{-1, 1}]`
[12]	Locator で動かす：	`Manipulate[ ray[v], {{v, {1, 1}}, Locator}]`
[13]	0 と z^2 を結ぶ線分：	`sq[z_] := ParametricPlot[`
	：	`t vect[z^2], {t, 0, 1}, PlotRange -> 5]`
[14]	たとえば…：	`sq[1 + 2 I]`
[15]	`Locator` で動かす：	`Manipulate[ sq[comp[v]], {{v, {2, 1}}, Locator}]`

問題 3.（もっと見やすく） [13] のベクトル　`t vect[z^2]`　をベクトルのリスト　`t {vect[z], vect[z^2]}`　に変えて，z と z^2 の位置関係が「絶対値は 2 乗，偏角は 2 倍」を満たしていることを確認せよ．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

テイラー級数，およびオイラーの等式 $e^{\pi i}=-1$ の可視化

いわゆるオイラーの公式 $e^{i\theta}=\cos\theta+i\sin\theta$ で $\theta=\pi$ としたものは**オイラーの等式**とよばれている．円周率 π，自然対数の底 e，虚数単位 i の豪華メンバーがかくも単純な関係式で結ばれるのだから，不思議なものである．

[16]	指数関数 e^z を定義：	`f[z_] := Exp[z];`
[17]	テイラー多項式を関数化：	`g[z_, n_] := Normal[Series[f[w], {w, 0, n}]]/. w -> z;`
[18]	$z=\pi i$ で確認：	`g[Pi I, 10]`
[19]	`Table` にしてみる：	`Table[g[Pi I, k], {k, 0, 10}]`
[20]	e^z の近似数列を線分で：	`seg[z_] := ListLinePlot[`
	結んだグラフを描く関数：	`Table[vect[g[z, k]], {k, 0, 20}],`
	（オプション 1）：	`PlotRange -> {{-4, 4}, {-4, 4}},`
	（オプション 2）：	`AspectRatio -> Automatic,`
	（オプション 3）：	`PlotMarkers -> Automatic];`
[21]	オイラーの等式を可視化：	`seg[Pi I]`
[22]	z を `Locator` で動かす：	`Manipulate[ seg[comp[v]], {{v, {2, 1}}, Locator}]`

問題 4.（その他の級数） [16][17][20][22] を応用して，$\sin z$, $1/(1-z)$ のテイラー級数が収束もしくは発散する様子を折れ線で「動的に」図示せよ．

複素関数の「グラフ」

[23]	関数・変数を定義：	`f[z_] := Sin[z]; z := x + y I`
[24]	実部のグラフを描く：	`Plot3D[ Re[f[z]], {x, 0, 4 Pi}, {y, 0, 3},`
	：	`BoxRatios -> Automatic]`
[25]	関数を定義：	`g[z_] := z^2 - 1; z := r Exp[I t]`
[26]	$z\mapsto\lvert g(z)\rvert$ の 3 次元グラフ：	`ParametricPlot3D[{Re[z], Im[z], Abs[g[z]]},`
	（半径 2 の円板上）：	`{r, 0, 2}, {t, 0, 2 Pi}, PlotRange -> All]`

問題 5.（複素正弦の虚部と絶対値） [23] を応用して，$z\mapsto \mathrm{Im}\,(\sin z)$，$z\mapsto \lvert\sin z\rvert$ のグラフを描け．

問題 6.（最大値の原理） [26] の出力結果にたいして，$g(z)=z^2-1$ の半径 2 の円板上における最大絶対値は，境界で実現されることを確認せよ．また，この $g(z)$ を e^z や $\sin z$ に変えても同様の結果が得られることを確認せよ．

複素関数による正方形の像

[27]	関数と変数を定義：	`f[z_] := z^2; z := x + y I;`
[28]	正方形の像：	`ParametricPlot[vect[f[z]], {x, 0, 1}, {y, 0, 1}]`
[29]	もとの正方形と比較：	`ParametricPlot[{vect[z], vect[f[z]]}, {x, 0, 1}, {y, 0, 1}]`
[30]	w 中心一辺 $2r$ の正方形：	`mapf[w_] := ParametricPlot[{vect[z], vect[f[z]]},`
	とその像を描く関数：	`{x, Re[w] - r, Re[w] + r}, {y, Im[w] - r, Im[w] + r},`
	：	`PlotRange -> 5]`
[31]	たとえば…：	`r = 0.5; mapf[1 + I]`
[32]	`Locator` で動かす：	`Manipulate[mapf[comp[v]], {{v, {1, 1}}, Locator}]`

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

問題 7. (等角性) [32] の結果をよく眺め，もとの正方形の網目の像は，直交する曲線族であることを確認せよ．さらに，もとの正方形の頂点のひとつが原点 $z=0$ になるとき，直角が開いて 180 度になること確認せよ.（必要なら r の値を大きくしてよい．また，`mapf` に `ImageSize -> 500` などのオプションを加えると大きな絵が描ける.）

問題 8. (その他の関数の像) [27][30][32] の関数 $f(z)$ を z^3, $1/z$, e^z などに変えて写像の様子を調べよ．これらの場合においても，網目の像の直交性は保たれることを確認せよ．

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 08-060123456.m のようにしてメールで提出する．
あて先：`comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）：08-060123456
本文：もし今日の感想・要望・質問などがあれば書いてください．できるだけ回答します．

おねがい：退出時，iMac をログアウトでなく**シャットダウン**してください．

研究

問題 9. (テイラー級数) [22] の結果に，原点と `g[z,0]` を結ぶ線分を加えるにはどうすればよいか？

問題 10. (複素関数による円板の像) [32] の正方形を w 中心半径 r の円板に変えよ．また，r も変数として `Manipulate` できるスライダーを加えよ．

問題 11. (グラフを並べる) [32] の結果において，もとの正方形と像の正方形を別々のグラフに並べて描くにはどうすればよいだろうか？（下の問題がヒントになる.）

問題 12. (もっと使いやすく) 次を実行せよ．その上で，同等の機能を `Locator` を用いて実現せよ．

```
L = 1.3 Pi; f[z_]:= Exp[z];
Manipulate[
       { ParametricPlot[ {vect[a + I t], vect[t + I b]},
            {t, -L, L}, PlotRange -> L]
        ,
         ParametricPlot[ {vect[f[a + I t]], vect[f[t + I b]]},
            {t, -L, L}, PlotRange -> 5] }
   , {{a, 1}, -L, L}, {{b, 1}, -L, L}]
```

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

確率実験とランダムウォーク

作成日： December 08, 2011　Version： 1.2

◎！注意！◎：問題だけを解くのではなく，[1][2]…等のコマンドも必ず入力し，結果を確かめること．ファイルの提出時もこれらの入力内容は消さずに残しておくこと．

必要な関数の準備

[1]	1 が 10 個の Table：	`tab = Table[1, {10}]`
[2]	Accumulate で**足し上げ**：	`Accumulate[tab]`
[3]	グラフを描く関数：	`gra[n_] := Plot[Sin[n x], {x, 0, Pi}]`
[4]	リストにする：	`gratab = Table[gra[n], {n, 1, 5}]`
[5]	**グラフを重ねあわせる**：	`Show[gratab, PlotRange -> All]`

乱数を作る関数

[6]	0 か 1 だけの**乱数**:	`x := RandomInteger[];`
[7]	たとえば…：	`Table[x, {10}]`
[8]	$\|y\| \leq 3$ の**整数値をとる乱数**：	`y := RandomInteger[{-3, 3}]`
[9]	たとえば…：	`Table[y, {10}]`
[10]	$0 \leq x \leq 1$ の**実数値乱数**：	`x := RandomReal[]; Table[x, {10}]`
[11]	$9 \leq x \leq 10$ の実数値乱数：	`x := RandomReal[{9, 10}]; Table[x, {10}]`
[12]	**リストから等確率で 20 個選ぶ**：	`RandomChoice[{a, b, c}, 20]`

問題 1. (遅延的定義でなきゃだめ) [10] は遅延的定義 `x :=` を用いている．これを即時的定義 `x =` にすると，結果はどうなるか？

問題 2. (株価っぽいグラフ) 株価 (share/stock) っぽいグラフを描こう．

(1) `x := RandomReal[{-1, 1}]` とする．`dif = Table[x, {100}]` にたいし，`shr = Accumulate[dif]]` を求めよ．

(2) `ListLinePlot[shr]` としてグラフを描け．

確率実験（コイン投げ）

コインを 10 回投げて (toss)，表が出た回数だけ金貨（ゴールド）がもらえるとする．このゲームをシミュレーションしてみよう．コインの表を 1，裏を 0 とする．

[13]	コインを 10 回投げる：	`toss := Table[RandomInteger[], {10}]`
[14]	たとえば…：	`toss`
[15]	もらえるゴールドをカウント：	`gold := Total[toss]`
[16]	たとえば…：	`gold`
[17]	100 人にやらせる：	`n = 100; trials = Table[gold, {n}]`
[18]	k ゴールドの人数をカウント：	`num[k_] := Count[trials, k]`
[19]	分布の折れ線グラフ：	`ListLinePlot[ Table[ num[k], {k, 0, 10}] ]`

問題 3. (ゲームの価格設定) さいころを 5 回投げて，((出た目の最大値) ＋ (出た目の最小値)) × 100 円をもらえる有料のゲームを作りたい．1000 人にこのゲームをやらせてみて，損益分岐価

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

格を自分なりに決定せよ．ただし，リストの総和・最大値・最小値を与える組み込み関数はそれぞれ `Total[...]`, `Max[...]`, `Min[...]` である．

1 次元ランダムウォーク

数直線の原点に立ち，コインを投げて表なら正の方向へ 1 移動，裏なら負の方向へ 1 移動，という操作を繰り返す．その様子をグラフ化してみよう．ただし，コインの表は 1，裏は −1 と考える．

[20]	−1 と 1 をランダムに選ぶ：	`x := RandomChoice[{-1, 1}]`
[21]	n 回選ぶ関数：	`p[n_] := Table[x, {n}]`
[22]	たとえば…：	`p[100]`
[23]	数直線での動きに直す：	`q[n_] := Accumulate[p[n]]`
[24]	たとえば…：	`q[100]`
[25]	動きをグラフ化：	`r[n_] := ListLinePlot[q[n]]`
[26]	たとえば…：	`r[10000]`
[27]	500 回を 10 人にやらせる：	`tab = Table[r[500], {10}]`
[28]	グラフを重ねる：	`Show[tab, PlotRange -> All]`

問題 4. (グラフの色分け) 重ねたグラフはとても見づらいので，10 人それぞれ別の色になるようにしよう．

(1) 上の [25] で，`r[n_, c_] := ListLinePlot[q[n], PlotStyle-> Hue[c/10]]` とせよ．

(2) さらに [27] で，`tab = Table[r[500, c], {c,1,10}]` とし， [28] を実行せよ．

2 次元ランダムウォーク

今度は 2 次元平面の原点からスタートし,「上下左右の方向に，それぞれ確率 1/4 で 1 移動」という操作を繰り返す．その様子をグラフにしてみよう．

[29]	上下左右の方向：	`dir = {{0, 1}, {0, -1}, {1, 0}, {-1, 0}};`
[30]	方向を等確率で選ぶ変数：	`x := RandomChoice[dir]`
[31]	方向を n 回選ぶ関数：	`p[n_] := Table[x, {n}]`
[32]	たとえば…：	`p[10]`
[33]	`p[n]` の足し合わせ関数；	`q[n_] := Accumulate[p[n]]`
[34]	折れ線プロット関数：	`r[n_, c_] := ListLinePlot[ q[n],`
	（オプション）：	`AspectRatio -> Automatic, PlotStyle -> Hue[c/10]];`
[35]	たとえば…：	`r[10000, 1]`
[36]	10 人にやらせる：	`tab = Table[r[5000, c], {c, 1, 10}]`
[37]	グラフを重ねる：	`Show[tab, PlotRange -> All]`

問題 5. (偏りのあるランダムウォーク) [29] の `dir` をうまく修正し，上下に進む確率がそれぞれ 1/3，左右に進む確率がそれぞれ 1/6 となるようにせよ．そのうえで [36][37] を実行せよ．

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 09-060123456.m のようにしてメールで提出する．
あて先：`comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）：09-060123456
本文：もし今日の感想・要望・質問などがあれば書いてください．できるだけ回答します．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

> **おねがい：**退出時，iMac をログアウトでなく**シャットダウン**してください．

研究：モンテカルロ法 (the Monte Carlo method)

乱数を用いて円周率の近似値を求めることができる：まず，xy 平面上に単位円板を含む正方形 $\{|x|, |y| \le 1\}$ をとる．この正方形内にランダムに点をうっていくと，単位円板の中に入っている点の割合は（円板の面積）÷（正方形の面積）$= \pi/4$ に近づくであろう．円板に入っているかどうかは原点との距離の 2 乗で判定できる．また，第一象限だけで実験すればよい．

[38]	区間 $[0, 1]$ に値をとる乱数：	`x := RandomReal[]; y := RandomReal[];`
[39]	原点からの距離の 2 乗：	`p[n_] := Table[x^2 + y^2, {n}]`
[40]	たとえば…：	`p[10]`
[41]	条件判定関数：	`condi[x_] := x <= 1;`
[42]	たとえば…：	`condi[5]`
[43]	condi をみたすものを数える：	`k[n_] := Length[Select[p[n], condi]]`
[44]	たとえば…：	`k[100]`
[45]	円周率を近似する：	`n = 10000; N[4 k[n]/n, 5]`

研究：3 次元ランダムウォーク



図 1: 3 次元ランダムウォーク

3 次元ランダムウォークも次のようにすれば描くことが出来る．

```
dir = {{0, 0, 1}, {0, 0, -1}, {1, 0, 0},
                    {-1, 0, 0},{0, -1, 0}, {0, 1, 0}};
  x := RandomChoice[dir]
```

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

```
p[n_] := Table[x, {n}]
q[n_] := Accumulate[p[n]]
r[n_, c_] := Graphics3D[{Hue[c/20], Line[q[n]]}, AspectRatio -> Automatic]
Show[Table[r[1000, c], {c, 1, 20}], PlotRange -> All]
```

研究：2 次元ランダムウォークいろいろ

問題 6. (六方格子上のランダムウォーク) 正 6 角形のタイルで埋め尽くされた平面を**六方格子** (hexagonal lattice) とよぶ．[29] の `dir` と [31] の `p[n_]` をうまく修正し，六方格子上のランダムウォークを作成せよ．そのうえで [36][37] を実行せよ．(HINT. `dir` 進んだら次は `-dir` 進む．)



図 2: 六方格子上のランダムウォーク

問題 7. (動くランダムウォーク) 次を実行すると `Manipulate` のグラフができる．スライダ横の＋（プラス）を押して，再生ボタンをおすとアニメーションになる．再生速度を遅くして，ランダムウォークの様子を観察せよ．

```
dir = {{0, 1}, {0, -1}, {1, 0}, {-1, 0}};
x := RandomChoice[dir];
p[n_] := Table[x, {n}]
n = 10000; q = Accumulate[p[n]];
qq[k_] := Table[q[[i]], {i, 1, k}]
r[k_] := ListLinePlot[qq[k], PlotStyle -> Thick];
total = ListLinePlot[qq[n], PlotStyle -> Red];
Manipulate[Show[{r[k], total}, PlotRange -> All,
  AspectRatio -> Automatic, ImageSize -> 500], {k, 1, n}]
```

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

ベクトル場と微分方程式

作成日： December 15, 2011 Version： 1.2

◎！注意！◎：問題だけを解くのではなく，[1][2]…等のコマンドも必ず入力し，結果を確かめること．ファイルの提出時もこれらの入力内容は消さずに残しておくこと．

2 次元ベクトル場

ベクトル場 (vector field) $\boldsymbol{v}: \mathbb{R}^n \to \mathbb{R}^n$ とは，$\mathbb{R}^n$ の各点 $\boldsymbol{p}$ に「速度ベクトル」 $\boldsymbol{v}(\boldsymbol{p}) \in \mathbb{R}^n$ を対応付けたものである．風をイメージするのが一番わかりやすい．また，時刻 $t \in \mathbb{R}$ でパラメーター付けされた曲線 $t \mapsto \boldsymbol{p}(t)$ が $\dfrac{d\boldsymbol{p}(t)}{dt} = \boldsymbol{v}(\boldsymbol{p}(t))$ をみたすとき，この曲線をベクトル場 $\boldsymbol{v}$ の**流線** (stream line) もしくは**積分曲線** (integral curve) と呼ぶ．

[1]	**ベクトル場**を描く：	`VectorPlot[{1, Cos[x]}, {x, -3, 3}, {y, -3, 3}]`
[2]	その**流線**（正弦曲線）：	`StreamPlot[{1, Cos[x]}, {x, -3, 3}, {y, -3, 3}]`
[3]	例その２：	`VectorPlot[{-y, x}, {x, -3, 3}, {y, -3, 3}]`
[4]	流線は円を描く：	`StreamPlot[{-y, x}, {x, -3, 3}, {y, -3, 3}]`

問題 1. 上と同じ範囲でベクトル場 $\boldsymbol{v} = (y, -\sin 2x - y)$ を `VectorPlot` および `StreamPlot` せよ．また，範囲を原点の周りに限定した場合（たとえば $|x|, |y| \le 0.3$），$\boldsymbol{v}$ は $\boldsymbol{u} = (y, -2x - y)$ にきわめて近いことを確認せよ．それはなぜか？

勾配ベクトル場

2 変数関数 $f = f(x, y)$ の**勾配ベクトル** (gradient vector) とは，ベクトル $\left(\dfrac{\partial f}{\partial x}, \dfrac{\partial f}{\partial y}\right)$ のことである．[1] たとえば f が海抜高度を表す関数であるとき，地面にボールを置くと，ボールは勾配ベクトルと「逆向きに」転がっていく．

[5]	平面上の関数：	`f = Cos[x] + y*Sin[x];`
[6]	3 次元グラフ：	`Plot3D[f, {x, -5, 5}, {y, -5, 5}]`
[7]	関数の等高線プロット：	`c = ContourPlot[f, {x, -5, 5}, {y, -5, 5}]`
[8]	**勾配ベクトル**：	`gradf = D[f, {{x, y}}]`
[9]	**勾配ベクトル場**：	`v = VectorPlot[gradf, {x, -5, 5}, {y, -5, 5}]`
[10]	その流線：	`s = StreamPlot[gradf, {x, -5, 5}, {y, -5, 5}]`
[11]	それらを重ねて表示：	`Show[c, v]`
[12]	：	`Show[c, s]`

問題 2. [5]—[12] を別の関数 $g = g(x, y)$ （自分が面白いと思うもの）に変えて実行せよ．

微分方程式を解く（厳密解と数値解）

Mathematica は（連立）微分方程式の厳密解を

`DSolve[{方程式, 初期条件, ...}, 求める関数名, 変数名]`

で与える．(解けないこともある．) 同様に，数値解は次の形であたえられる：

`NDSolve[{方程式, 初期条件, ...}, 求める関数名, 変数名とその変域]`

[1] これを $\operatorname{grad} f = \operatorname{grad} f(x, y)$ などとあらわす．

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

[13]	**厳密解**を求める（正弦曲線）：	`sol = DSolve[{y'[x] == Cos[x], y[0] == 1}, y[x], x]`
[14]	解を関数 yy に代入：	`yy[x_] = y[x] /. sol[[1]]`
[15]	値をチェック：	`yy[Pi]`
[16]	グラフを描く：	`Plot[yy[x], {x, 0, 2 Pi}]`
[17]	円の連立微分方程式：	`sol2 = DSolve[{x'[t] == -y[t], y'[t] == x[t],`
	：	`x[0] == 1, y[0] == 0}, {x[t], y[t]}, t]`
[18]	解をベクトル関数に代入：	`p[t_] = {x[t], y[t]} /. sol2[[1]]`
[19]	グラフ：	`ParametricPlot[p[t], {t, -Pi, Pi}]`

数値解を求めた場合，結果に `InterpolatingFunction[...]` というのが出てくるが，気にしなくてもよい．とにかく，厳密解と同様に扱うことができる．

[20]	**数値解**を求める：	`sol = NDSolve[{y'[x] == y[x]*(1 - y[x]),`
	（ロジスティック曲線）：	`y[0] == 0.02}, y[x], {x, 0, 10}]`
[21]	解を関数 yy に代入：	`yy[x_] = y[x] /. sol[[1]]`
[22]	値をチェック：	`yy[4]`
[23]	グラフを描く：	`Plot[yy[x], {x, 0, 10}]`
[24]	線形連立微分方程式：	`sol = NDSolve[ {x'[t] == 2 y[t], y'[t] == -2 x[t] - y[t],`
	：	`x[0] == 1, y[0] == 0}, {x[t], y[t]}, {t, 0, 10}];`
[25]	解をベクトル関数に代入：	`p[t_] = {x[t], y[t]} /. sol[[1]]`
[26]	グラフ：	`ParametricPlot[p[t], {t, 0, 10}, PlotRange -> All]`

問題 3. [24] の方程式は $\begin{pmatrix} x' \\ y' \end{pmatrix} = A \begin{pmatrix} x \\ y \end{pmatrix}$，$A = \begin{pmatrix} 0 & 2 \\ -2 & -1 \end{pmatrix}$ の形をしている．[2] この A を便宜的に「定義行列」と呼ぼう．ベクトル場 $A\begin{pmatrix} x \\ y \end{pmatrix}$ を `StreamPlot` し，[25] の `p[t]` がその流線であることを確認せよ．このような流線はとくに，微分方程式の**解曲線**もしくは**解軌道**とよばれる．

問題 4. （初期値を変える） `Manipulate` の `Locator` を用いて，[24] の初期値を変えたときの解軌道の変化がわかるようにしたい．

(1) 次を入力せよ：

```
str[{qx_, qy_}] := (sol = NDSolve[{x'[t] == 2 y[t], y'[t] == -2 x[t] - y[t],
                              x[0] == qx, y[0] == qy}, {x[t], y[t]}, {t, 0, 10}];
                    p[t_] = {x[t], y[t]} /. sol[[1]];
                    ParametricPlot[p[t], {t, 0, 10}, PlotRange -> 5]
                    )
Manipulate[str[q], {{q, {1, 1}}, Locator}]
```

(2) [24] の定義行列 $A = \begin{pmatrix} 0 & 2 \\ -2 & -1 \end{pmatrix}$ を次のものに変えて，(1) と同じことをせよ．また，それぞれの行列について問題 3 のように `StreamPlot` し，解軌道の大域的な様子を観察せよ．

（ア）$\begin{pmatrix} -1 & 2 \\ -1 & 1 \end{pmatrix}$ （イ）$\begin{pmatrix} 4 & 3 \\ -2 & 11 \end{pmatrix}$ （ウ）$\begin{pmatrix} 8 & -9 \\ 6 & -13 \end{pmatrix}$ （エ）$\begin{pmatrix} 2 & 9 \\ -1 & 8 \end{pmatrix}$

[2] 実は厳密解が存在し，$\boldsymbol{p}(t) = \boldsymbol{p}(0)\exp(tA)$ と書ける．ただし，$\exp(X) = I + X + X^2/2! + X^3/3! + \cdots$．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

課題の提出： 以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 10-060123456.m のようにしてメールで提出する．
あて先： `comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）： 10-060123456
本文： もし今日の感想・要望・質問などがあれば書いてください．できるだけ回答します．

おねがい： 退出時，iMac をログアウトでなく**シャットダウン**してください．

研究：ロトカ-ヴォルテラ方程式

時間 t でのウサギ（被食者）とキツネ（捕食者）の個体数をそれぞれ $x = x(t)$, $y = y(t)$ としたとき，これらの関数は次の**ロトカ-ヴォルテラ方程式** (Lotka-Volterra Equations) でモデル化できる．

$$x' = x(\alpha - \beta y), \quad y' = -y(\gamma - \delta x) \qquad (\alpha, \beta, \gamma, \delta > 0 \text{ は定数})$$

(1) [24][25][26] を応用して，ロトカ-ヴォルテラ方程式を定数 $\alpha = \gamma = 1$, $\beta = 0.04$, $\delta = 0.02$, 初期値 $x(0) = 100, y(0) = 10$, 変域 $0 \le t \le 20$ として解き，解軌道を xy 平面に図示せよ．

(2) (1) の結果（解軌道が閉曲線のようになる）は何を意味するか考えよ．

(3) (1) の `ParametricPlot` をただの `Plot` に変えて結果を比較せよ．

(4) (3) を修正し，`Manipulate` を用いて $\delta = 0.02$ を $0 \le d \le 0.1$ の範囲で変えられるようにせよ.（Hint. 問題 4(1) を応用する．この場合 `Locator` は不要.）

研究：ローレンツ・アトラクター

次の方程式は**ローレンツ方程式** (Lorenz equation) と呼ばれる．解曲線が集積する部分は**ローレンツ・アトラクター** (Lorenz attractor) と呼ばれ，極めて複雑な図形となる．

```
sol = NDSolve[
        {x'[t] == -10 x[t] + 10 y[t],
         y'[t] == - x[t] z[t] + 28 x[t] - y[t],
         z'[t] == x[t] y[t] - (8/3) z[t],
         x[0] == 1, y[0] == 0, z[0] == 0}, {x[t], y[t], z[t]}, {t, 0, 50}];
p[t_] := {x[t], y[t], z[t]} /. sol[[1]];
ParametricPlot3D[p[t], {t, 0, 50}, PlotRange -> All]
```

研究：3 次元勾配ベクトル場

2 次元勾配ベクトル場の 3 次元版．計算量が多いため動作は重い．

[27]	3 変数関数：	`f = x^2 + y^2 + z^2;`
[28]	勾配ベクトル：	`gradf = D[f, {{x, y, z}}]`
[29]	ベクトル場：	`v = VectorPlot3D[gradf, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]`
[30]	等高線：	`c = ContourPlot3D[f, {x, -2, 2}, {y, -2, 2}, {z, -2, 2},`
[31]	（オプション）：	`Mesh -> None, ContourStyle -> Opacity[0.5, Green]]`
[32]	グラフを重ねる：	`Show[v, c]`

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

数列と素数のふるまい

作成日： December 22, 2011　Version： 1.2

Tips： 下の [1] のように数列や関数の「値」を定義した場合，明示的に上書きしない限り *Mathematica* は内部で値を記憶しつづける．悪影響を避けるためには，

- 数列・関数・変数を頻繁に Clear する，もしくは
- 数列や関数で同じ文字を繰り返し使わない．

◎！注意！◎：問題だけを解くのではなく，[1][2]⋯等のコマンドも必ず入力し，結果を確かめること．ファイルの提出時もこれらの入力内容は消さずに残しておくこと．

漸化式による数列の定義

[1]	フィボナッチ数列：	`Fib[0] = 1; Fib[1] = 1;`
[2]	：	`Fib[n_] := (Fib[n] = Fib[n - 1] + Fib[n - 2])`
[3]	10 項目まで列挙：	`tab = Table[Fib[n], {n, 0, 10}]`
[4]	折れ線プロット：	`ListLinePlot[tab]`

問題 1. (大学入試問題) 数列 $\{a_n\}$ を次のように定める．

$$a_1 = 2,\ a_2 = 3,\quad a_{n+2} - a_n = 4\quad (n = 1, 2, 3, \ldots)$$

このとき，$a_3 = \boxed{}$，$a_4 = \boxed{}$，$a_5 = \boxed{}$, $a_6 = \boxed{}$ であり，$a_{40} = \boxed{}$ である．また，$\displaystyle\sum_{k=1}^{40} a_k = \boxed{}$ である．(Hint. [1][2][3] と同様に 40 項目まで列挙，リストの和は `Total[]`．)

(01 センター本試)

`Rsolve` で漸化式を微分方程式っぽく解く（いわゆる差分方程式）

[5]	2 項間漸化式：	`sol = RSolve[{a[n + 1] == 2 a[n] + 1, a[0] == 1}, a[n], n]`
[6]	解を数列 aa に代入：	`aa[n_] := a[n] /. sol[[1]]`
[7]	10 項目までリストにする：	`tab = Table[aa[n], {n, 0, 10}]`

問題 2. (再フィボナッチ) [5][6][7] を応用しフィボナッチ数列の漸化式を解き，10 項目までのリストを作れ．

問題 3. `RSolve` を用いて**問題 1** の漸化式を解き，10 項目までのリストを作れ．

関数と `Nest` による数列の生成

[8]	関数 f の 5 回反復：	`Nest[f, x, 5]`
[9]	関数を定義：	`f[x_] := x + 10;`
[10]	Nest させる：	`Nest[f, x, 5]`
[11]	数列を定義：	`a[n_] := Nest[f, 1, n];`
[12]	表にする：	`Table[a[n], {n, 1, 10}]`
[13]	表を作る組み込み関数：	`b[n_] := NestList[f, 1, n]`
[14]	たとえば⋯：	`b[10]`
[15]	条件関数を決めて⋯：	`condi[k_] := (k < 200)`
[16]	その条件で Nest を止める：	`bb[n_] := NestWhileList[f, n, condi]`
[17]	たとえば⋯：	`bb[24]`

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

If を用いた場合分け

条件に応じて値を決める関数は組み込み関数 `If` を用いて作ることができる．

`If`(条件式，`True` のときの値，`False` のときの値)

[18]	たとえば…：	`g[x_] := If[x < 0, Sin[x], x]`
[19]	グラフ：	`Plot[g[x], {x, -20, 5}]`

問題 4.（ヘビサイド関数） $x < 0$ のとき 0，$x \geq 0$ のとき 1 となるような関数を作れ．

コラッツの問題

与えられた自然数 n にたいし，偶数ならば 2 で割る，奇数ならば $3n+1$ にする，という操作を繰り返すと，かならず $\cdots \mapsto 1 \mapsto 4 \mapsto 2 \mapsto 1$ というサイクルに落ち着く，というのが**コラッツの予想**（the Collatz conjecture, 未解決）である．[1]

以下では，`NestWhileList`，`If`，リストの長さを与える組み込み関数 `Length` などを用いてこの問題を観察する．

[20]	偶数を判定する組み込み関数：	`EvenQ[10]`
[21]	Collatz 関数を `If` で作成：	`col[n_] := If[EvenQ[n], n/2, 3 n + 1];`
[22]	たとえば…：	`col[17]`
[23]	「1 でない」ことの条件関数：	`condi[k_] := (k > 1);`
[24]	操作が終了するまでのリスト：	`colList[n_] := NestWhileList[col, n, condi]`
[25]	たとえば…：	`colList[9]`
[26]	グラフ化：	`ListLinePlot[colList[9]]`
[27]	`n` を動かした表：	`tab = Table[colList[n], {n, 1, 10}]`
[28]	`TableForm` で比較：	`tab // TableForm`
[29]	操作が終了するのにかかる時間：	`len[n_] := Length[colList[n]]`
[30]	`n` を動かした表：	`tab2 = Table[len[n], {n, 1, 100}];`
[31]	グラフ化：	`ListPlot[tab2, Filling -> Axis, PlotRange -> All]`

問題 5. 操作が終了するまでの時間が長い `n` を上のグラフから 1 から 100 までの間で適当に探し，その `colList` をグラフ化せよ．[2]

素数の分布

正の実数 x 以下の素数の個数を，整数論では習慣的に $\pi(x)$ で表す． 関数としての $\pi(x)$ は階段状の関数であるが，いくつかの近似公式が知られている．たとえば次のものが有名である：

- ガウスの近似：$\pi(x) \sim \dfrac{x}{\log x}$ （ただし $f(x) \sim g(x) \; :\Longleftrightarrow \; \dfrac{f(x)}{g(x)} \to 1 \;\; (x \to \infty)$ ）
- チェビシェフの近似：$\pi(x) \sim \displaystyle\int_0^x \frac{dt}{\log t}$ （この関数は *Mathematica* の `LogIntegral[x]`）

[32]	10000 番目の素数：	`Prime[10000]`
[33]	105000 以下の素数の数：	`PrimePi[105000]`

[1] その他，角谷の問題，$3n+1$ 予想，シラキューズ予想など，さまざまな呼び名がある．

[2] ちなみに 1 億以下では `n = 63728127` が最長らしい．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

[34]	1 万以下の素数のカウント：	`list1 = Table[PrimePi[n], {n, 2, 10000}];`
[35]	：	`gr1 = ListPlot[list1, PlotStyle -> Red]`
[36]	ガウスの近似：	`list2 = Table[n/Log[n], {n, 2, 10000}];`
[37]	：	`gr2 = ListPlot[list2, PlotStyle -> Green]`
[38]	チェビシェフの近似：	`list3 = Table[LogIntegral[n], {n, 2, 10000}];`
[39]	：	`gr3 = ListPlot[list3, PlotStyle -> Blue]`
[40]	グラフを比較：	`Show[gr1, gr2, gr3]`

問題 6. (近似良し悪し) すでに計算させたリストを用いて，ガウスの近似とチェビシェフの近似の誤差を比較するグラフを描け．(Hint リストはベクトルでもある.)

> **課題の提出：**以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 11-060123456.m のようにしてメールで提出する．
> **あて先：**`comp0-2011@math.nagoya-u.ac.jp`
> **件名（半角で！）：**11-060123456
> **本文：**もし今日の感想・要望・質問などがあれば書いてください．できるだけ回答します．

> **おねがい：**退出時，iMac をログアウトでなく**シャットダウン**してください．

研究：双子素数を探す

整数のペア $(p, p+2)$ がともに素数であるとき，これらを**双子素数** (twin prime) とよぶ．双子素数は無限個あると予想されているが，未解決である．ここでは素数のリストから `Select` 関数をもちいて双子素数を列挙してみよう．

[41]	条件関数：	`condi[n_] := (n > 4);`
[42]	リストから条件で選択：	`Select[{1, 4, 6, 7}, condi]`
[43]	双子素数用の条件設定：	`condi[{p1_, p2_}] := (p2 - p1 == 2);`
[44]	素数のペアのリスト：	`tab = Table[{Prime[n], Prime[n + 1]}, {n, 1, 1000}];`
[45]	双子素数を選ぶ：	`twins = Select[tab, condi];`
[46]	結構たくさんある：	`twins // TableForm`
[47]	比率を求める：	`twinsRatio = N[Length[twins]/Length[tab]]`

問題 7. ($4k+1$ vs $4k+3$) 2 以外の素数は一般に $4k+1$ もしくは $4k+3$ の形をしている．上の双子素数を見つける操作を参考に，$4k+1$ および $4k+3$ の形の素数をそれぞれ取り出してリストにせよ．また，それぞれの比率を求めよ．ただし，`n` を 4 で割ったあまりは組み込み関数 `Mod[n, 4]` で与えられる．

問題 8. (三つ子素数 prime triplet) 整数の三つ組 $(p, p+2, p+6)$ または $(p, p+4, p+6)$ がすべて素数からなるとき，これを三つ子素数とよぶ．三つ子素数を列挙せよ．三つ子素数も無限個あると予想されているが，未解決である．ただし,「$x=0$ かつ $y=0$」は `(x == 0) && (y == 0)`,「$x=0$ または $y=0$」は `(x == 0)||(y == 0)` のように表す．

研究

問題 9. (偽フィボナッチ) 数列 $b_n = \dfrac{\phi^{n+1}}{\sqrt{5}}$ $(n \geq 0)$ はフィボナッチ数列を極めてよく近似することを確認せよ．ただし，$\phi = (\sqrt{5}+1)/2$ は黄金比であり，*Mathematica* では `GoldenRatio` と表される．

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

問題 10.（ロジスティック・カオス） Nest を用いて数列のカオス的なふるまいを観察しよう．以下を入力すると，関数 $f_c(x) := cx(1-x)$ により $1/2 \mapsto f_c(1/2) \mapsto f_c(f_c(1/2)) \mapsto \cdots$ で与えられる数列が c に応じてカオス的に変化することが観察できる．

```
f[c_][x_] := c x (1 - x);
lis[n_, c_] := NestList[f[c], 0.5, n]
gra[n_, c_] := ListLinePlot[lis[n, c], PlotRange -> {{0, n}, {-2, 2}}]
Manipulate[gra[20, c], {c, -5, 5}]
```

問題 11.（バンプ関数） 多様体の微積分で重要な C^∞ 関数である，**バンプ関数** (bump function) のグラフを描こう．

(1) If をもちいて，$a(x) := \begin{cases} 0 & (x \le 0) \\ e^{-1/x^2} & (x > 0) \end{cases}$ を定義し，グラフを描け．

(2) $b(x) := \dfrac{a(x)}{a(x) + a(1-x)}$ のグラフを描け．

(3) 適当に $\epsilon > 0$ をとり，$c(x) := b(x/\epsilon + 2)b(-x/\epsilon + 2)$ のグラフをかけ．

問題 12.（続・素数の分布） 歴史上有名な $\pi(x)$ の近似式として，

- ルジャンドルの近似：$\pi(x) \sim \dfrac{x}{\log x - 1.08366}$
- リーマンの近似：$\pi(x) \sim 1 + \displaystyle\sum_{n=1}^{\infty} \left(\frac{(\log x)^n}{n \cdot n! \cdot \zeta(n+1)} \right)$ （ただし $\zeta(s) = \displaystyle\sum_{n=1}^{\infty} \frac{1}{n^s}$ ）

が知られている．これらの近似式と $\pi(x)$ をグラフ上で比較せよ．また，ガウス，チェビシェフのものとも比較せよ．ただし，リーマンの近似式は組み込み関数 RiemannR[x] で与えられる．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

セル・オートマトン

作成日： January 12, 2012　Version : 1.1

ArrayPlot による行列の表現

ArrayPlot は行列の各要素を正方形のセル（もしくはピクセル）で表現する．ただし，ここでいう「行列」とは,「リストのリスト」のこと．

[1]	0 と 1 によるランダムな 10 × 20 行列：	`m = Table[RandomInteger[], {10}, {20}];`
[2]	TableForm で表現：	`TableForm[m]`
[3]	ArrayPlot で表現：	`ArrayPlot[m]`
[4]	0 と 1 の間のランダムな実数の場合：	`m = Table[RandomReal[], {10}, {20}];`
[5]	：	`TableForm[m]`
[6]	：	`ArrayPlot[m]`

問題 1. (カラーにする) 0 と 1 の間の実数 x にたいし，Hue[x] とすれば色相環[1]に基づいた「色」が指定できる．たとえば Hue[0] は Red，Hue[1/3] は Green，Hue[2/3] は Blue といった具合である．

いま行列を m = Table[Hue[RandomReal[]], {10}, {20}]; と定めて，これを TableForm および ArrayPlot で表せ．また，行列のサイズを適当に変化させよ．

パスカルの三角形

パスカルの三角形 (Pascal's triangle) とは，いわゆる 2 項係数 ${}_n\mathrm{C}_k = \dfrac{n!}{k!(n-k)!}$ を並べたものである．これは $(x+1)^n$ を展開したときの x^k の係数に等しく，さらに関係式

$$ {}_{n+1}\mathrm{C}_k = {}_n\mathrm{C}_{k-1} + {}_n\mathrm{C}_k $$

が成り立つのであった．これらの数を整数で割ったあまりがどのように分布するか，ArrayPlot で観察してみよう．ただし，p を q で割ったあまりは組み込み関数 Mod[p,q] で与えられる．

[7]	2 項係数を定義：	`c[n_, k_] := n!/(k!*(n - k)!)`
[8]	一覧表にする：	`pas = Table[c[n, k], {n, 1, 10}, {k, 0, n}];`
[9]	TableForm で表現：	`TableForm[pas]`
[10]	n=100 のとき，mod 2 の表：	`pas2 = Table[Mod[c[n, k], 2], {n, 1, 100}, {k, 0, n}];`
[11]	ArrayPlot する：	`ArrayPlot[pas2]`

問題 2. (カラーにする) pascal[j_] = Table[Hue[Mod[c[n, k], j]/j], {n, 1, 100}, {k, 0, n}]; とおき，Manipulate を用いて，pascal[j] を ArrayPlot したものが j が 2 から 100 まで変化できるようにせよ.（j は整数値のみを取ることに注意！）

If による条件分岐

[12]	If で不連続関数を場合分けで定義：	`f[x_] := If[x < 0, -1, 1];`
[13]	そのグラフ：	`Plot[f[x], {x, -2, 2}]`
[14]	If を繰り返し，さらに場合分け：	`g[x_] := If[x < 0, -1, If[x < 1, 1/2, 1]];`
[15]	：	`Plot[g[x], {x, -2, 2}]`

[1]赤～緑～青～赤という色の変化が円状に配置されたもの．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

問題 3.（周期性をもった数列の作成） `If` と `Mod[n,3]` を用いて，π, e, i, π, e, i, $\pi, \ldots$ と続く数列 `a[n]` を `a[n_] := If[...]` の形で定義し，100 項目までのリストを作成せよ．

セル・オートマトンを自作する

周期的境界をもつ**セル・オートマトン** (cellular automaton) を作成しよう．（詳細は講義で説明する．）

[16]	初期状態の作成（100 個の 0）：	`n = 101; c0 = Table[0, {n - 1}];`
[17]	50 番目に 1 を挿入：	`c = Insert[c0, 1, (n - 1)/2];`
[18]	`ArrayPlot` で表現：	`ArrayPlot[{c}]`
[19]	ルール 150 を定める関数：	`f[x_, y_, z_] := Mod[x + y + z, 2];`
[20]	i 番目のセルに f で定まる：	`rule[c_, i_] :=`
	ルールを適用する関数：	`If[i == 1, f[ c[[n]], c[[1]], c[[2]] ],`
	：	`If[i == n, f[ c[[n-1]], c[[n]], c[[1]] ],`
	：	`f[ c[[i-1]], c[[i]], c[[i+1]] ]`
	（ふたつの `If` の閉じ括弧→）：	`] ]`
[21]	次のセル状態を作る関数：	`next[c_] := Table[rule[c, i], {i, 1, n}];`
[22]	もとの c に適用：	`c = next[c]; ArrayPlot[{c}]`
[23]	もう一回：	`c = next[c]; ArrayPlot[{c}]`
[24]	さらにもう一回：	`c = next[c]; ArrayPlot[{c}]`
[25]	その続きをタテに並べる：	`ArrayPlot[NestList[next, c, 200]]`

問題 4.（その他のルール） 以上を応用して，次のように変化させよ．

(1) [18] の初期状態から，[20] を変形させた関数 `f[x_, y_, z_] := Mod[x + y + z + y z, 2];` によるルールを適用した場合，0 ステップ目から 200 ステップ目までの状態を [26] のように `ArrayPlot` せよ．（これはルール 30 となる．）

(2) (1) の初期状態を 0 と 1 からなるランダムな列に変えた場合はどうか？

(3) ほかにもルールを自作して，どのようなパターンが得られるか確認せよ．（状態数を 0, 1, 2 などに増やすのも面白い．`Hue` を使ってカラーにしても面白い．この場合，一旦 200 ステップ目までの `NestList` を作ってから，各要素が 0 と 1 の間の値になるように調整し，さらに各要素の `Hue` からなる行列を `Table` で作るとよい．）

組み込み関数によるセル・オートマトン

Mathematica の開発者 S. Wolfram はもともと，セル・オートマトン理論のエキスパートである．もちろん *Mathematica* の中にも，セル・オートマトンを生成するための組み込み関数が準備されている．

[26]	初期状態を作成：	`c0 := Table[0, {100}]; c := Insert[c0, 1, 50];`
[27]	c にルール 30 を 200 回適用：	`b := CellularAutomaton[30, c, 200];`
[28]	`ArrayPlot` で表現：	`ArrayPlot[b]`

問題 5.（いろんなルール） `Manipulate` をもちいて，初期状態 c にルール 0 から 255 までを適用したセル・オートマトンが図示できるようにせよ．（ルールの番号は整数だけをわたるようにすること．）

担当教員：川平 友規 研究室：A441 Email: kawahira@math.nagoya-u.ac.jp

課題の提出： 以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 12-060123456.m のようにしてメールで提出する．
あて先： `comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）： 12-060123456
本文： もし今日の感想・要望・質問などがあれば書いてください．できるだけ回答します．

おねがい： 退出時，iMac をログアウトでなく**シャットダウン**してください．

図 1: イモ貝の殻（出典：Wikipedia）

研究

問題 6. (アニメーション) `Manipulate` を用いて，ルール 30 によるセル・オートマトンのアニメーションを作成せよ.（これにはふたつの方法が考えられる．一列のセルが時間に応じて変化するものと，列が下に伸びていって時間発展の様子がわかるように変化するもの．いずれでもよい.）

問題 7. (いろんな境界条件) 上であつかった `n` 個のセルからなるセル・オートマトンでは，境界がなく，セルは**周期的**（すなわち円環状にセルが並んでいる）と考えた．これは列の両端に仮想的なセル `c[[0]] = c[[n]]`，`c[[n + 1]] = c[[1]]` を考えていることに対応する．そのほか，境界条件として次のものがある．

- **固定的：** 仮想的に `c[[0]]` と `c[[n + 1]]` を定数として設定する．
- **反射的：** 仮想的に `c[[0]] = c[[2]]`, `c[[n + 1]] = c[[n-1]]` として設定する．

これらの場合，セル・オートマトンがどのように変化するか確かめよ.（`n` が大きすぎると，変化がわかりづらいかもしれない.）

担当教員：川平 友規　研究室：A441　　Email: kawahira@math.nagoya-u.ac.jp

フラクタル

作成日： January 19, 2012　Version：1.2

Nest 以外のループ処理（Do，For，While）

関数や漸化式の反復は Nest を用いて実現できた．それ以外にも，*Mathematica* には一般的なプログラミング言語で使われるようなループ処理（反復処理）用の組み込み関数が備わっている．

以下はすべて，1 から 5 までをタテに並べて表示するためのプログラムである：

[1] Do の使い方： `Do[Print[i], {i, 1, 5}]`
[2] While の使い方： `i = 1; While[i < 6, Print[i]; i = i + 1]`
[3] For の使い方： `For[i = 1, i < 6, i = i + 1, Print[i]]`

これらは次のように解釈する：

- Do は,「Print[i] を i = 1 から i = 5 まで（1 刻みで）動かして実行せよ」．i の刻み幅を 0.5 にしたければ，Do[Print[i], {i, 1, 5, 0.5}] のように指定できる．
- While は,「まず i = 1 とせよ．つぎに条件 i < 6 が満たされている限り，Print[i] と i = i + 1 を実行せよ」
- For は,「まず i = 1 とせよ．つぎに条件 i < 6 が満たされている限り，i = i + 1 と Print[i] を交互に実行せよ」[1]

問題 1. (プログラムを読む)　『x = 1; Do[x = x * i, {i, 1, 5}]; x』と『x = 1; Do[x = x * i, {i, 5, 1, -1}]; x』はともに 5! を計算していることを納得せよ．また，同等のプログラムを While を用いて書け．

リストをほどく，置換する

フタクタル図形を描くにあたって，必要なリストの操作を理解しておこう．

リストのほどき方（平坦化）

[4] 3 重のリスト： `list = { {{a, a}, {b, b}}, {{c, c}, {d, d}} };`
[5] 全レベル平坦化： `Flatten[list]`
[6] 1 レベル平坦化： `Flatten[list, 1]`

リストの和集合

[7] {a , b} と {c , d} の和集合： `Join[{a , b}, {c , d}]`

リストの要素の置換

[8] 3 番目の項を xxx に置換： `ReplacePart[{a, b, c, d, e}, 3 -> xxx]`

リストの各項に関数を施す

[9] リストの各要素に関数 f を適用： `Map[f, {a, b, c, d}]`

[1] C 言語など他のプログラミング言語と比べて，For の使い方が若干ややこしい．プログラムの読みやすさを考えると，*Mathematica* では使わないほうがいいかもしれない．

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

コッホ曲線

複素平面上で，0 と 1 を結ぶ線分からスタートし，次のような操作を限りなく続けて得られる「極限」の図形が**コッホ曲線** (Koch curve) である.（「極限」の数学的な意味，詳しい構成方法については授業で解説する.）この図形を（近似的に）作図しよう.



関数の準備． いくつか必要な関数を定義しておく.

[10]	複素数→ベクトル：	`vect[z_] = {Re[z], Im[z]};`
[11]	たとえば：	`vect[1 + I]`
[12]	複素数のリスト→ベクトルのリスト：	`vectorize[list_] := Map[vect, list]`
[13]	たとえば：	`vectorize[{-1, I, -I, 1}]`
[14]	複素数のリストを線分で結ぶ：	`complexLLP[list_] :=`
	：	`ListLinePlot[vectorize[list],`
	：	`AspectRatio -> Automatic,`
	：	`Axes -> None, ImageSize -> 600];`
[15]	たとえば：	`complexLLP[{-1, I, -I, 1}]`

問題 2. **(Map の効用)** `vectorize2[list_] := vect[list]` としたとき，`vectorize2[{-1, I, -I, 1}]` はどうなるか？（これでは `ListLinePlot` を適用しても思うような絵にならない！）

コッホ曲線を描く．

[16]	定数を決める：	`s = Exp[Pi/6 I]/Sqrt[3.0] ;`
[17]	リストの置換用関数：	`rep[a_, b_] :=`
	：	`{a + (b - a)/3, a + s*(b - a), a + (b - a)*(2/3), b};`
[18]	最初の線分：	`koch = {0.0, 1.0};`
[19]	次の線分（ループ）：	`koch = (ko = koch;`
	：	`Do[`
	：	`ko = ReplacePart[ko, i -> rep[ ko[[i-1]], ko[[i]] ]],`
	：	`{i, Length[koch], 2, -1}];`
	：	`Flatten[ko]);`
	(図示する)：	`complexLLP[koch]`

問題 3. **(手動で反復)** [19] を繰り返し実行して（再入力は不要だよ，念のため），コッホ曲線に近づいていく様子を観察せよ.

問題 4. [19] を `While` を用いて書き直せ.（時間が足りなそうなら，省略して次にすすむこと.）

担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

IFS によるコッホ曲線

定数 $s = \dfrac{e^{\pi i/6}}{\sqrt{3}}$ を用いて関数 $T_i : \mathbb{C} \to \mathbb{C}\ (i = 1, 2)$ を

$$T_1(z) = -s(z-1), \quad T_2(z) = -\bar{s}z + 1$$

で定めるとき，コッホ曲線は

- $K_0 = (0, 1, s$ を頂点とする三角形$)$
- $K_{n+1} = T_1(K_n) \cup T_2(K_n)$

として得られる図形の列 $\{K_n\}_{n \geq 0}$ の「極限」とも考えられる．



[20]	さっきの定数：	`s = Exp[Pi/6 I]/Sqrt[3.0] ;`
[21]	関数その 1 ：	`T1[z_] := -s (z - 1);`
[22]	関数その 2 ：	`T2[z_] := -Conjugate[s] z + 1;`
[23]	K_0：	`ver = {{0, 1, s}}`
[24]	K_n から K_{n+1} へ：	`ver = Join[Map[T1, ver], Map[T2, ver]];`
	（複素数→ベクトル）：	`triangles = Map[vectorize, ver];`
	図形の描画：	`Graphics[Polygon[triangles], ImageSize -> 600]`

問題 5. (手動で反復) [24] を繰り返し実行して，コッホ曲線に近づいていく様子を観察せよ．

問題 6. (シェルピンスキー・ガスケット) 定数 $\alpha = e^{\pi i/3}$ を用いて関数 $S_i : \mathbb{C} \to \mathbb{C}$ を

$$S_1(z) = z/2, \quad S_2(z) = \frac{1}{2}(z-1)+1, \quad S_3(z) = \frac{1}{2}(z-\alpha)+\alpha$$

で定めるとき，

- $K_0 = (0, 1, \alpha$ を頂点とする三角形$)$
- $K_{n+1} = S_1(K_n) \cup S_2(K_n) \cup S_3(K_n)$

として得られる図形の列 $\{K_n\}_{n \geq 0}$ の「極限」は**シェルピンスキー・ガスケット**（Sierpinski's gasket）と呼ばれる．上の [20]—[24] を応用して，$K_1, K_2, \ldots$ を描画せよ．



担当教員：川平 友規　研究室：A441　Email: kawahira@math.nagoya-u.ac.jp

課題の提出：以上の問題を解いたノートブック（.nb）ファイルをパッケージ形式（.m）として保存し，ファイル名を 13-060123456.m のようにしてメールで提出する．
あて先：`comp0-2011@math.nagoya-u.ac.jp`
件名（半角で！）：13-060123456
本文：もし今日の感想・要望・質問などがあれば書いてください．できるだけ回答します．

おねがい：退出時，iMac をログアウトでなく**シャットダウン**してください．

研究

問題 7. (ドラゴン) 次のようにして得られる曲線の「極限」は**ハイウェイ・ドラゴン** (Heighway's dragon) と呼ばれる．これを描画せよ．



問題 8. (カーペット) 正方形から中央の正方形を次のように除いていって得られる「極限」は**シェルピンスキー・カーペット** (Sierpinski's carpet) と呼ばれる．IFS を用いてこれを描画せよ．



問題 9. (ペンタ君？) 正五角形からスタートし，次のような操作を繰り返して得られる図形の「極限」は**ペンタクン** (pentakun) と呼ばれる．IFS を用いてこれを描画せよ．(Hint: 1 の 5 乗根を頂点にもつ 5 角形からスタート．これらの頂点を中心に複素平面を縮小率 $(3-\sqrt{5})/2$ で縮小する写像を用いる．)

担当教員：川平 友規　研究室：A441 Email: kawahira@math.nagoya-u.ac.jp



問題 10.（カラーでペンタ君） 上のようにランダムなカラーでペンタクンを描け．（HINT: 色のついた多角形は `Graphics[{Red, Polygon[..]}]` のように指定する．）

参考． 今回紹介したような図形は**フラクタル** (fractal) と呼ばれる．これらの共通する性質は**自己相似性** (self-similarity) である．このような図形には整数でない次元を定義することができる．たとえば…

- コッホ曲線：$\frac{\log 4}{\log 3} \approx 1.262$ 次元
- シェルピンスキー・ガスケット：$\frac{\log 3}{\log 2} \approx 1.585$ 次元
- ハイウェイ・ドラゴン：2 次元（平面を部分的に埋め尽くす）
- シェルピンスキー・カーペット：$\frac{\log 8}{\log 3} \approx 1.893$ 次元
- ペンタクン：$\frac{\log 5}{\log(3-\sqrt{5})-\log 2} \approx 1.672$ 次元